
Getting Started
    

This guide is an introduction to Nebulex,
a local and distributed caching toolkit for Elixir. Nebulex API is pretty much
inspired by Ecto, taking advantage of
its simplicity, flexibility and pluggable architecture. In the same way
as Ecto, developers can provide their own cache (adapter) implementations.
In this guide, we're going to learn some basics about Nebulex, such as insert,
retrieve and destroy cache entries.

  
    
  
  Adding Nebulex to an application


Let's start creating a new Elixir application by running this command:
mix new blog --sup
The --sup option ensures that this application has
a supervision tree,
which will be needed by Nebulex later on.
To add Nebulex to this application, there are a few steps that we need to take.
The first step will be adding Nebulex to our mix.exs file, which we'll do by
changing the deps definition in that file to this:
defp deps do
  [
    {:nebulex, "~> 2.6"},
    {:shards, "~> 1.0"},      #=> When using :shards as backend
    {:decorator, "~> 1.4"},   #=> When using Caching Annotations
    {:telemetry, "~> 1.0"}    #=> When using the Telemetry events (Nebulex stats)
  ]
end
In order to give more flexibility and loading only needed dependencies, Nebulex
makes all its dependencies as optional. For example:
	For intensive workloads, you may want to use :shards as the backend for
the local adapter and having partitioned tables. In such a case, you have
to add :shards to the dependency list.

	For enabling the usage of
declarative annotation-based caching via decorators,
you have to add :decorator to the dependency list.

	For enabling Telemetry events to be dispatched when using Nebulex,
you have to add :telemetry to the dependency list.
See telemetry guide.

	If you want to use an external adapter (e.g: Cachex or Redis adapter), you
have to add the adapter dependency too.


To install these dependencies, we will run this command:
mix deps.get
We now need to define a Cache and setup some configuration for Nebulex so that
we can perform actions on a cache from within the application's code.
We can set up this configuration by running this command:
mix nbx.gen.cache -c Blog.Cache
This command will generate the configuration required to use the cache. The
first bit of configuration is in config/config.exs:
config :blog, Blog.Cache,
  # When using :shards as backend
  # backend: :shards,
  # GC interval for pushing new generation: 12 hrs
  gc_interval: :timer.hours(12),
  # Max 1 million entries in cache
  max_size: 1_000_000,
  # Max 2 GB of memory
  allocated_memory: 2_000_000_000,
  # GC min timeout: 10 sec
  gc_cleanup_min_timeout: :timer.seconds(10),
  # GC max timeout: 10 min
  gc_cleanup_max_timeout: :timer.minutes(10)
Assuming we will use :shards as backend, can add uncomment the first line in
the config
config :blog, Blog.Cache,
  # When using :shards as backend
  backend: :shards,
  # GC interval for pushing new generation: 12 hrs
  gc_interval: :timer.hours(12),
  # Max 1 million entries in cache
  max_size: 1_000_000,
  # Max 2 GB of memory
  allocated_memory: 2_000_000_000,
  # GC min timeout: 10 sec
  gc_cleanup_min_timeout: :timer.seconds(10),
  # GC max timeout: 10 min
  gc_cleanup_max_timeout: :timer.minutes(10)
By default, partitions: option is set to System.schedulers_online().

NOTE: For more information about the provided options, see the adapter's
documentation.
And the Blog.Cache module is defined in lib/blog/cache.ex by our
mix nbx.gen.cache command:
defmodule Blog.Cache do
  use Nebulex.Cache,
    otp_app: :blog,
    adapter: Nebulex.Adapters.Local
end
This module is what we'll be using to interact with the cache. It uses the
Nebulex.Cache module and it expects the :otp_app as option. The otp_app
tells Nebulex which Elixir application it can look for cache configuration in.
In this case, we've specified that it is the :blog application where Nebulex
can find that configuration and so Nebulex will use the configuration that was
set up in config/config.exs.
The final piece of configuration is to setup the Blog.Cache as a
supervisor within the application's supervision tree, which we can do in
lib/blog/application.ex, inside the start/2 function:
def start(_type, _args) do
  children = [
    Blog.Cache
  ]

  ...
This piece of configuration will start the Nebulex process which receives and
executes our application's commands. Without it, we wouldn't be able to use
the cache at all!
We've now configured our application so that it's able to execute commands
against our cache.
IMPORTANT: Make sure the cache is put in first place within the children
list, or at least before the process or processes using it. Otherwise, there
could be race conditions causing Nebulex.RegistryLookupError errors;
processes attempting to use the cache and this one hasn't been even
started.

  
    
  
  Inserting entries


We can insert a new entries into our blog cache with this code:
iex> user = %{id: 1, first_name: "Galileo", last_name: "Galilei"}
iex> Blog.Cache.put(user[:id], user, ttl: :timer.hours(1))
:ok
To insert the data into our cache, we call put on Blog.Cache. This function
tells Nebulex that we want to insert a new key/value entry into the cache
corresponding Blog.Cache.
It is also possible to insert multiple entries at once:
iex> users = %{
...>   1 => %{id: 1, first_name: "Galileo", last_name: "Galilei"},
...>   2 => %{id: 2, first_name: "Charles", last_name: "Darwin"},
...>   3 => %{id: 3, first_name: "Albert", last_name: "Einstein"}
...> }
iex> Blog.Cache.put_all(users)
:ok
The given entries can be a map or a Key/Value tuple list.


  
    
  
  Inserting new entries and replacing existing ones


As we saw previously, put creates a new entry in cache if it doesn't exist,
or overrides it if it does exist (including the :ttl). However, there might
be circumstances where we want to set the entry only if it doesn't exit or the
other way around, this is where put_new and replace functions come in.
Let's try put_new and put_new! functions:
iex> new_user = %{id: 4, first_name: "John", last_name: "Doe"}
iex> Blog.Cache.put_new(new_user.id, new_user, ttl: 900)
true

iex> Blog.Cache.put_new(new_user.id, new_user)
false

# same as previous one but raises `Nebulex.KeyAlreadyExistsError`
iex> Blog.Cache.put_new!(new_user.id, new_user)
Now replace and replace! functions:
iex> existing_user = %{id: 5, first_name: "John", last_name: "Doe2"}
iex> Blog.Cache.replace(existing_user.id, existing_user)
false

iex> Blog.Cache.put_new(existing_user.id, existing_user)
true

iex> Blog.Cache.replace(existing_user.id, existing_user, ttl: 900)
true

# same as previous one but raises `KeyError`
iex> Blog.Cache.replace!(100, existing_user)
It is also possible to insert multiple new entries at once:
iex> new_users = %{
...>   6 => %{id: 6, first_name: "Isaac", last_name: "Newton"},
...>   7 => %{id: 7, first_name: "Marie", last_name: "Curie"}
...> }
iex> Blog.Cache.put_new_all(new_users)
true

# none of the entries is inserted if at least one key already exists
iex> Blog.Cache.put_new_all(new_users)
false

  
    
  
  Retrieving entries


Let’s start off with fetching data by the key, which is the most basic and
common operation to retrieve data from a cache.
iex> Blog.Cache.get(1)
_user_1

iex> for key <- 1..3 do
...>   user = Blog.Cache.get(key)
...>   user.first_name
...> end
["Galileo", "Charles", "Albert"]
There is a function has_key? to check if a key exist in cache:
iex> Blog.Cache.has_key?(1)
true

iex> Blog.Cache.has_key?(10)
false
Retrieving multiple entries
iex> Blog.Cache.get_all([1, 2, 3])
_users

  
    
  
  Updating entries


Nebulex provides update and get_and_update functions to update an
entry value based on current one, for example:
iex> initial = %{id: 1, first_name: "", last_name: ""}

# using `get_and_update`
iex> Blog.Cache.get_and_update(1, fn v ->
...>   if v, do: {v, %{v | first_name: "X"}}, else: {v, initial}
...> iex> end)
{_old, _updated}

# using `update`
iex> Blog.Cache.update(1, initial, &(%{&1 | first_name: "Y"}))
_updated

  
    
  
  Counters


The function incr is provided to increment or decrement a counter; by default,
a counter is initialized to 0. Let's see how counters works:
# by default, the counter is incremented by 1
iex> Blog.Cache.incr(:my_counter)
1

# but we can also provide a custom increment value
iex> Blog.Cache.incr(:my_counter, 5)
6

# to decrement the counter, just pass a negative value
iex> Blog.Cache.incr(:my_counter, -5)
1

  
    
  
  Deleting entries


We’ve now covered inserting, reading and updating entries. Now let's see how to
delete an entry using Nebulex.
iex> Blog.Cache.delete(1)
:ok

  
    
  
  Take


This is another way not only for deleting an entry but also for retrieving it
before its delete it:
iex> Blog.Cache.take(1)
_entry

# returns `nil` if `key` doesn't exist
iex> Blog.Cache.take("nonexistent")
nil

# same as previous one but raises `KeyError`
iex> Blog.Cache.take!("nonexistent")

  
    
  
  Info


The last thing we’ll cover in this guide is how to retrieve information about
cached objects or the cache itself.

  
    
  
  Remaining TTL


iex> Blog.Cache.ttl(1)
_remaining_ttl

iex> Blog.Cache.ttl("nonexistent")
nil

  
    
  
  Query and/or Stream entries


Nebulex provides functions to fetch, count, delete, or stream all entries from
cache matching the given query.

  
    
  
  Fetch all entries from cache matching the given query


# by default, returns all keys
iex> Blog.Cache.all()
_all_entries

# fetch all entries and return the keys
iex> Blog.Cache.all(nil, return: :key)
_keys

# built-in queries in `Nebulex.Adapters.Local` adapter
iex> Blog.Cache.all(nil)
iex> Blog.Cache.all(:unexpired)
iex> Blog.Cache.all(:expired)

# if we are using `Nebulex.Adapters.Local` adapter, the stored entry
# is a tuple `{:entry, key, value, touched, ttl}`, then the match spec
# could be something like:
iex> spec = [{{:_, :"$1", :"$2", :_, :_}, [{:>, :"$2", 10}], [{{:"$1", :"$2"}}]}]
iex> Blog.Cache.all(spec)
_all_matched

# using Ex2ms
iex> import Ex2ms
iex> spec =
...>   fun do
...>     {_, key, value, _, _} when value > 10 -> {key, value}
...>   end
iex> Blog.Cache.all(spec)
_all_matched

  
    
  
  Count all entries from cache matching the given query


For example, to get the total number of cached objects (cache size):
iex> Blog.Cache.count_all()
_num_cached_entries
By default, since none query is given to count_all/2, all entries
in cache match.

In the same way as all/2, you can pass a query to count only the matched
entries:
# using Ex2ms
iex> import Ex2ms
iex> spec =
...>   fun do
...>     {_, value, _, _} when rem(value, 2) == 0 -> true
...>   end
iex> Blog.Cache.count_all(spec)
_num_of_matched_entries
The previous example assumes you are using the built-in local adapter.

Also, if you are using the built-in local adapter, you can use the queries
:expired and :unexpired too, like so:
iex> expired_entries = Blog.Cache.count_all(:expired)
iex> unexpired_entries = Blog.Cache.count_all(:unexpired)

  
    
  
  Delete all entries from cache matching the given query


Similar to count_all/2, Nebulex provides delete_all/2 to not only count
the matched entries but also remove them from the cache at once, in one single
execution.
The first example is flushing the cache, delete all cached entries (which is
the default behavior when none query is provided):
iex> Blog.Cache.delete_all()
_num_of_removed_entries
And just like count_all/2, you can also provide a custom query to delete only
the matched entries, or if you are using the built-in local adapter you can also
use the queries :expired and :unexpired. For example:
iex> expired_entries = Blog.Cache.delete_all(:expired)
iex> unexpired_entries = Blog.Cache.delete_all(:unexpired)

# using Ex2ms
iex> import Ex2ms
iex> spec =
...>   fun do
...>     {_, value, _, _} when rem(value, 2) == 0 -> true
...>   end
iex> Blog.Cache.delete_all(spec)
_num_of_matched_entries
These examples assumes you are using the built-in local adapter.


  
    
  
  Stream all entries from cache matching the given query


Similar to all/2 but returns a lazy enumerable that emits all entries from the
cache matching the provided query.
If the query is nil, then all entries in cache match and are returned when the
stream is evaluated; based on the :return option.
iex> Blog.Cache.stream()
iex> Blog.Cache.stream(nil, page_size: 100, return: :value)
iex> Blog.Cache.stream(nil, page_size: 100, return: :entry)

# using `Nebulex.Adapters.Local` adapter
iex> spec = [{{:"$1", :"$2", :_, :_}, [{:>, :"$2", 10}], [{{:"$1", :"$2"}}]}]
iex> Blog.Cache.stream(spec)
_all_matched

# using Ex2ms
iex> import Ex2ms
iex> spec =
...>   fun do
...>     {key, value, _, _} when value > 10 -> {key, value}
...>   end
iex> Blog.Cache.stream(spec)
_all_matched

  
    
  
  Partitioned Cache


Nebulex provides the adapter Nebulex.Adapters.Partitioned, which allows to
set up a partitioned cache topology.
Let's set up the partitioned cache by using the mix task mix nbx.gen.cache:
mix nbx.gen.cache -c Blog.PartitionedCache -a Nebulex.Adapters.Partitioned
As we saw previously, this command will generate the cache in
lib/bolg/partitioned_cache.ex (in this case using the partitioned adapter)
module along with the initial configuration in config/config.exs.
The cache:
defmodule Blog.PartitionedCache do
  use Nebulex.Cache,
    otp_app: :blog,
    adapter: Nebulex.Adapters.Partitioned,
    primary_storage_adapter: Nebulex.Adapters.Local
end
And the config:
config :blog, Blog.PartitionedCache,
  primary: [
    # When using :shards as backend
    backend: :shards,
    # GC interval for pushing new generation: 12 hrs
    gc_interval: :timer.hours(12),
    # Max 1 million entries in cache
    max_size: 1_000_000,
    # Max 2 GB of memory
    allocated_memory: 2_000_000_000,
    # GC min timeout: 10 sec
    gc_cleanup_min_timeout: :timer.seconds(10),
    # GC max timeout: 10 min
    gc_cleanup_max_timeout: :timer.minutes(10)
  ]
And remember to add the new cache Blog.PartitionedCache to your application's
supervision tree (such as we did it previously):
def start(_type, _args) do
  children = [
    Blog.Cache,
    Blog.PartitionedCache
  ]

  ...
Now we are ready to start using our partitioned cache!

  
    
  
  Timeout option


The Nebulex.Adapters.Partitioned supports :timeout option, it is a value in
milliseconds for the command that will be executed.
iex> Blog.PartitionedCache.get("foo", timeout: 10)
_value

# if the timeout is exceeded, then the current process will exit
iex> Blog.PartitionedCache.put("foo", "bar", timeout: 10)
# ** (EXIT) time out
To learn more about how partitioned cache works, please check
Nebulex.Adapters.Partitioned documentation, and also it is recommended see the
partitioned cache example

  
    
  
  Multilevel Cache


Nebulex also provides the adapter Nebulex.Adapters.Multilevel, which allows to
setup a multi-level caching hierarchy.
First, let's set up the multi-level cache by using the mix task
mix nbx.gen.cache:
mix nbx.gen.cache -c Blog.NearCache -a Nebulex.Adapters.Multilevel
By default, the command generates a 2-level near-cache topology. The first
level or L1 using the built-in local adapter, and the second one or L2
using the built-in partitioned adapter.
The generated cache module lib/blog/near_cache.ex:
defmodule Blog.NearCache do
  use Nebulex.Cache,
    otp_app: :blog,
    adapter: Nebulex.Adapters.Multilevel

  ## Cache Levels

  # Default auto-generated L1 cache (local)
  defmodule L1 do
    use Nebulex.Cache,
      otp_app: :blog,
      adapter: Nebulex.Adapters.Local
  end

  # Default auto-generated L2 cache (partitioned cache)
  defmodule L2 do
    use Nebulex.Cache,
      otp_app: :blog,
      adapter: Nebulex.Adapters.Partitioned
  end

  ## TODO: Add, remove or modify the auto-generated cache levels above
end
And the configuration (config/config.exs):
config :blog, Blog.NearCache,
  model: :inclusive,
  levels: [
    # Default auto-generated L1 cache (local)
    {
      Blog.NearCache.L1,
      # GC interval for pushing new generation: 12 hrs
      gc_interval: :timer.hours(12),
      # Max 1 million entries in cache
      max_size: 1_000_000
    },
    # Default auto-generated L2 cache (partitioned cache)
    {
      Blog.NearCache.L2,
      primary: [
        # GC interval for pushing new generation: 12 hrs
        gc_interval: :timer.hours(12),
        # Max 1 million entries in cache
        max_size: 1_000_000
      ]
    }
  ]
Remember you can add backend: :shards to use Shards as backend.

Finally, add the new cache Blog.NearCache to your application's supervision
tree (such as we did it previously):
def start(_type, _args) do
  children = [
    Blog.Cache,
    Blog.PartitionedCache,
    Blog.NearCache
  ]

  ...
Let's try it out!
iex> Blog.NearCache.put("foo", "bar", ttl: :timer.hours(1))
"bar"

iex> Blog.NearCache.get("foo")
"bar"
To learn more about how multilevel-cache works, please check
Nebulex.Adapters.Multilevel documentation, and also it is recommended see the
near cache example

  
    
  
  Next


	Cache Usage Patterns via Nebulex.Caching -
Annotations-based DSL to implement different cache usage patterns.



  

    
Cache Usage Patterns via Nebulex.Caching
    

There are several common access patterns when using a cache. Nebulex
supports most of these patterns by means of Nebulex.Caching.
Most of the following documentation about caching patterns it based on
EHCache Docs


  
    
  
  Cache-aside


With the cache-aside pattern, application code uses the cache directly.
This means that application code which accesses the system-of-record (SoR)
should consult the cache first, and if the cache contains the data, then return
the data directly from the cache, bypassing the SoR. Otherwise, the application
code must fetch the data from the system-of-record, store the data in the cache,
and then return it. When data is written, the cache must be updated along with
the system-of-record.

  
    
  
  Reading values


if value = MyCache.get(key) do
  value
else
  value = SoR.get(key) # maybe Ecto?
  :ok = MyCache.put(key, value)
  value
end

  
    
  
  Writing values


:ok = MyCache.put(key, value)
SoR.insert(key, value) # maybe Ecto?
As you may have noticed, this is the default behavior for most of the caches,
we have to interact directly with the cache as well as the SoR (most likely the
Database).

  
    
  
  Cache-as-SoR


The cache-as-SoR pattern implies using the cache as though it were the
primary system-of-record (SoR).
The pattern delegates SoR reading and writing activities to the cache, so that
application code is (at least directly) absolved of this responsibility.
To implement the cache-as-SoR pattern, use a combination of the following
read and write patterns:
	Read-through

	Write-through


Advantages of using the cache-as-SoR pattern are:
	Less cluttered application code (improved maintainability through centralized
SoR read/write operations)

	Choice of write-through or write-behind strategies on a per-cache basis

	Allows the cache to solve the thundering-herd problem


A disadvantage of using the cache-as-SoR pattern is:
	Less directly visible code-path

But how to get all this out-of-box? This is where declarative annotation-based
caching comes in. Nebulex provides a set of annotation to abstract most of the
logic behind Read-through and Write-through patterns and make the
implementation extremely easy. But let's go over these patterns more in detail
and how to implement them by using Nebulex annotations.

  
    
  
  Read-through


Under the read-through pattern, the cache is configured with a loader component
that knows how to load data from the system-of-record (SoR).
When the cache is asked for the value associated with a given key and such an
entry does not exist within the cache, the cache invokes the loader to retrieve
the value from the SoR, then caches the value, then returns it to the caller.
The next time the cache is asked for the value for the same key it can be
returned from the cache without using the loader (unless the entry has been
evicted or expired).
This pattern can be easily implemented using cache decorator as follows:
defmodule MyApp.Example do
  use Nebulex.Caching

  alias MyApp.Cache

  @ttl :timer.hours(1)

  @decorate cacheable(cache: Cache, key: name)
  def get_by_name(name) do
    # your logic (the loader to retrieve the value from the SoR)
  end

  @decorate cacheable(cache: Cache, key: age, opts: [ttl: @ttl])
  def get_by_age(age) do
    # your logic (the loader to retrieve the value from the SoR)
  end

  @decorate cacheable(cache: Cache)
  def all(query) do
    # your logic (the loader to retrieve the value from the SoR)
  end
end
As you can see, the loader to retrieve the value from the system-of-record (SoR)
is the function logic itself.

  
    
  
  Write-through


Under the write-through pattern, the cache is configured with a writer component
that knows how to write data to the system-of-record (SoR).
When the cache is asked to store a value for a key, the cache invokes the writer
to store the value in the SoR, as well as updating (or deleting) the cache.
This pattern can be implemented using defevict or defupdatable. When the
data is written to the system-of-record (SoR), you can update the cached value
associated with the given key using defupdatable, or just delete it using
defevict.
defmodule MyApp.Example do
  use Nebulex.Caching

  alias MyApp.Cache

  # When the data is written to the SoR, it is updated in the cache
  @decorate cache_put(cache: Cache, key: something)
  def update(something) do
    # Write data to the SoR (most likely the Database)
  end

  # When the data is written to the SoR, it is deleted (evicted) from the cache
  @decorate cache_evict(cache: Cache, key: something)
  def update_something(something) do
    # Write data to the SoR (most likely the Database)
  end
end
As you can see, the logic to write data to the system-of-record (SoR) is the
function logic itself.


  

    
Telemetry
    

This guide is not focused on explaining :telemetry itself, how it works, how
to configure it, and so on. Instead, we will show you how to instrument and
report on Cache Telemetry events in your application when using Nebulex.
For more information about :telemetry, you can check the
documentation, or the Phoenix Telemetry
guide is also recommended.

  
    
  
  Telemetry Events


Many Elixir libraries (including Nebulex) are already using the :telemetry
package as a way to give users more insight into the behavior of their
applications, by emitting events at key moments in the application lifecycle.

  
    
  
  Nebulex built-in events


The following events are emitted by all Nebulex caches:
	[:nebulex, :cache, :init] - it is dispatched whenever a cache starts.
	Measurement: %{system_time: System.monotonic_time()}
	Metadata: %{cache: atom, opts: [term]}




  
    
  
  Adapter-specific events


Nebulex currently suggests the adapters to dispatch the following Telemetry
events:
	[:my_app, :cache, :command, :start] - Dispatched by the underlying cache
adapter before an adapter callback is executed.
	Measurement: %{system_time: System.monotonic_time()}
	Metadata: %{adapter_meta: map, function_name: atom, args: [term]}


	[:my_app, :cache, :command, :stop] - Dispatched by the underlying cache
adapter after an adapter callback has been successfully executed.
	Measurement: %{duration: native_time}

	Metadata:
%{
  adapter_meta: map,
  function_name: atom,
  args: [term],
  result: term
}



	[:my_app, :cache, :command, :exception] - Dispatched by the underlying
cache adapter when an exception is raised while the adapter callback is
executed.
	Measurement: %{duration: native_time}

	Metadata:
%{
  adapter_meta: map,
  function_name: atom,
  args: [term],
  kind: :error | :exit | :throw,
  reason: term,
  stacktrace: term
}





  
    
  
  Nebulex Metrics


Assuming you have defined the cache MyApp.Cache with the default
:telemetry_prefix, using Telemetry.Metrics, you can define a
counter metric, which counts how many cache commands were completed:
Telemetry.Metrics.counter("my_app.cache.command.stop.duration")
or you could use a distribution metric to see how many commands were completed
in particular time buckets:
Telemetry.Metrics.distribution(
  "my_app.cache.command.stop.duration",
  buckets: [100, 200, 300]
)
So far, these metrics are only helpful to be able to see just the total number
of executed cache commands. What if you wanted to see the average command
duration, minimum and maximum, or percentiles, but aggregated per command
or callback name? In this case, one could define a summary metric like so:
Telemetry.Metrics.summary(
  "my_app.cache.command.stop.duration",
  unit: {:native, :millisecond},
  tags: [:function_name]
)
As it is described above in the "Adapter-specific events" section, the event
includes the invoked callback name into the metadata as :function_name, then
we can add it to the metric's tags.

  
    
  
  Extracting tag values from adapter's metadata


Let's add another metric for the command event, this time to group by cache,
adapter, and function_name (adapter's callback):
Telemetry.Metrics.summary(
  "my_app.cache.command.stop.duration",
  unit: {:native, :millisecond},
  tags: [:cache, :adapter, :function_name],
  tag_values:
    &Map.merge(&1, %{
      cache: &1.adapter_meta.cache,
      adapter: &1.adapter_meta.cache.__adapter__()
    })
)
We've introduced the :tag_values option here, because we need to perform a
transformation on the event metadata in order to get to the values we need.

  
    
  
  Cache Stats


Each adapter is responsible for providing stats by implementing
Nebulex.Adapter.Stats behaviour. However, Nebulex provides a simple default
implementation using Erlang counters, which is used by
the built-in local adapter. The local adapter uses
Nebulex.Telemetry.StatsHandler to aggregate the stats and keep
them updated, therefore, it requires the Telemetry events are dispatched
by the adapter, otherwise, it won't work properly.
Furthermore, when the :stats option is enabled, we can use Telemetry for
emitting the current stat values.
First of all, make sure you have added :telemetry, :telemetry_metrics, and
:telemetry_poller packages as dependencies to your mix.exs file.
Let's define out cache module:
defmodule MyApp.Cache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Local
end
Make sure the :stats option is set to true, for example in the
configuration:
config :my_app, MyApp.Cache,
  stats: true,
  backend: :shards,
  gc_interval: :timer.hours(12),
  max_size: 1_000_000,
  gc_cleanup_min_timeout: :timer.seconds(10),
  gc_cleanup_max_timeout: :timer.minutes(10)
Create your Telemetry supervisor at lib/my_app/telemetry.ex:
# lib/my_app/telemetry.ex
defmodule MyApp.Telemetry do
  use Supervisor
  import Telemetry.Metrics

  def start_link(arg) do
    Supervisor.start_link(__MODULE__, arg, name: __MODULE__)
  end

  def init(_arg) do
    children = [
      # Configure `:telemetry_poller` for reporting the cache stats
      {:telemetry_poller, measurements: periodic_measurements(), period: 10_000},

      # For example, we use the console reporter, but you can change it.
      # See `:telemetry_metrics` for for information.
      {Telemetry.Metrics.ConsoleReporter, metrics: metrics()}
    ]

    Supervisor.init(children, strategy: :one_for_one)
  end

  defp metrics do
    [
      # Nebulex Stats Metrics
      last_value("my_app.cache.stats.hits", tags: [:cache]),
      last_value("my_app.cache.stats.misses", tags: [:cache]),
      last_value("my_app.cache.stats.writes", tags: [:cache]),
      last_value("my_app.cache.stats.updates", tags: [:cache]),
      last_value("my_app.cache.stats.evictions", tags: [:cache]),
      last_value("my_app.cache.stats.expirations", tags: [:cache])
    ]
  end

  defp periodic_measurements do
    [
      {MyApp.Cache, :dispatch_stats, []}
    ]
  end
end
Then add to your main application's supervision tree
(usually in lib/my_app/application.ex):
children = [
  MyApp.Cache,
  MyApp.Telemetry,
  ...
]
Now start an IEx session and call the server:
iex(1)> MyApp.Cache.get 1
nil
iex(2)> MyApp.Cache.put 1, 1, ttl: 10
:ok
iex(3)> MyApp.Cache.get 1
1
iex(4)> MyApp.Cache.put 2, 2
:ok
iex(5)> MyApp.Cache.delete 2
:ok
iex(6)> Process.sleep(20)
:ok
iex(7)> MyApp.Cache.get 1
nil
iex(2)> MyApp.Cache.replace 1, 11
true
and you should see something like the following output:
[Telemetry.Metrics.ConsoleReporter] Got new event!
Event name: my_app.cache.stats
All measurements: %{evictions: 2, expirations: 1, hits: 1, misses: 2, updates: 1, writes: 2}
All metadata: %{cache: MyApp.Cache}

Metric measurement: :hits (last_value)
With value: 1
Tag values: %{cache: MyApp.Cache}

Metric measurement: :misses (last_value)
With value: 2
Tag values: %{cache: MyApp.Cache}

Metric measurement: :writes (last_value)
With value: 2
Tag values: %{cache: MyApp.Cache}

Metric measurement: :updates (last_value)
With value: 1
Tag values: %{cache: MyApp.Cache}

Metric measurement: :evictions (last_value)
With value: 2
Tag values: %{cache: MyApp.Cache}

Metric measurement: :expirations (last_value)
With value: 1
Tag values: %{cache: MyApp.Cache}

  
    
  
  Custom metrics


In the same way, for instance, you can add another periodic measurement for
reporting the cache size:
defmodule MyApp.Cache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Local

  def dispatch_cache_size do
    :telemetry.execute(
      [:my_app, :cache, :size],
      %{value: size()},
      %{cache: __MODULE__, node: node()}
    )
  end
end
Now let's add a new periodic measurement to invoke dispatch_cache_size()
through :telemetry_poller:
defp periodic_measurements do
  [
    {MyApp.Cache, :dispatch_stats, [[metadata: %{node: node()}]]},
    {MyApp.Cache, :dispatch_cache_size, []}
  ]
end
Notice the node name was added to the metadata so we can use it in the
metric's tags.

Metrics:
defp metrics do
  [
    # Nebulex Stats Metrics
    last_value("my_app.cache.stats.hits", tags: [:cache, :node]),
    last_value("my_app.cache.stats.misses", tags: [:cache, :node]),
    last_value("my_app.cache.stats.writes", tags: [:cache, :node]),
    last_value("my_app.cache.stats.updates", tags: [:cache, :node]),
    last_value("my_app.cache.stats.evictions", tags: [:cache, :node]),
    last_value("my_app.cache.stats.expirations", tags: [:cache, :node]),

    # Nebulex custom Metrics
    last_value("my_app.cache.size.value", tags: [:cache, :node])
  ]
end
If you start an IEx session like previously, you should see the new metric too:
[Telemetry.Metrics.ConsoleReporter] Got new event!
Event name: my_app.cache.stats
All measurements: %{evictions: 0, expirations: 0, hits: 0, misses: 0, updates: 0, writes: 0}
All metadata: %{cache: MyApp.Cache, node: :nonode@nohost}

Metric measurement: :hits (last_value)
With value: 0
Tag values: %{cache: MyApp.Cache, node: :nonode@nohost}

Metric measurement: :misses (last_value)
With value: 0
Tag values: %{cache: MyApp.Cache, node: :nonode@nohost}

Metric measurement: :writes (last_value)
With value: 0
Tag values: %{cache: MyApp.Cache, node: :nonode@nohost}

Metric measurement: :updates (last_value)
With value: 0
Tag values: %{cache: MyApp.Cache, node: :nonode@nohost}

Metric measurement: :evictions (last_value)
With value: 0
Tag values: %{cache: MyApp.Cache, node: :nonode@nohost}

Metric measurement: :expirations (last_value)
With value: 0
Tag values: %{cache: MyApp.Cache, node: :nonode@nohost}

[Telemetry.Metrics.ConsoleReporter] Got new event!
Event name: my_app.cache.size
All measurements: %{value: 0}
All metadata: %{cache: MyApp.Cache, node: :nonode@nohost}

Metric measurement: :value (last_value)
With value: 0
Tag values: %{cache: MyApp.Cache, node: :nonode@nohost}

  
    
  
  Multi-level Cache Stats


When using the multi-level adapter the returned stats measurements may look
a bit different, because the multi-level adapter works as a wrapper for the
configured cache levels, so the returned measurements are grouped and
consolidated by level. For example, suppose you have the cache:
defmodule MyApp.Multilevel do
  use Nebulex.Cache,
    otp_app: :nebulex,
    adapter: Nebulex.Adapters.Multilevel

  defmodule L1 do
    use Nebulex.Cache,
      otp_app: :nebulex,
      adapter: Nebulex.Adapters.Local
  end

  defmodule L2 do
    use Nebulex.Cache,
      otp_app: :nebulex,
      adapter: Nebulex.Adapters.Partitioned
  end
end
Then, when you run MyApp.Multilevel.stats() you get something like:
%Nebulex.Stats{
  measurements: %{
    l1: %{evictions: 0, expirations: 0, hits: 0, misses: 0, updates: 0, writes: 0},
    l2: %{evictions: 0, expirations: 0, hits: 0, misses: 0, updates: 0, writes: 0}
  },
  metadata: %{
    l1: %{
      cache: NMyApp.Multilevel.L1,
      started_at: ~U[2021-01-10 13:06:04.075084Z]
    },
    l2: %{
      cache: MyApp.Multilevel.L2.Primary,
      started_at: ~U[2021-01-10 13:06:04.089888Z]
    },
    cache: MyApp.Multilevel,
    started_at: ~U[2021-01-10 13:06:04.066750Z]
  }
}
As you can see, the measurements map has the stats grouped by level, every key
is an atom specifying the level and the value is a map with the stats and/or
measurements for that level. Based on that, you could define the Telemetry
metrics in this way:
[
  # L1 metrics
  last_value("my_app.cache.stats.l1.hits",
    event_name: "my_app.cache.stats",
    measurement: &get_in(&1, [:l1, :hits]),
    tags: [:cache]
  ),
  last_value("my_app.cache.stats.l1.misses",
    event_name: "my_app.cache.stats",
    measurement: &get_in(&1, [:l1, :misses]),
    tags: [:cache]
  ),
  last_value("my_app.cache.stats.l1.writes",
    event_name: "my_app.cache.stats",
    measurement: &get_in(&1, [:l1, :writes]),
    tags: [:cache]
  ),
  last_value("my_app.cache.stats.l1.updates",
    event_name: "my_app.cache.stats",
    measurement: &get_in(&1, [:l1, :updates]),
    tags: [:cache]
  ),
  last_value("my_app.cache.stats.l1.evictions",
    event_name: "my_app.cache.stats",
    measurement: &get_in(&1, [:l1, :evictions]),
    tags: [:cache]
  ),
  last_value("my_app.cache.stats.l1.expirations",
    event_name: "my_app.cache.stats",
    measurement: &get_in(&1, [:l1, :expirations]),
    tags: [:cache]
  ),

  # L2 metrics
  last_value("my_app.cache.stats.l2.hits",
    event_name: "my_app.cache.stats",
    measurement: &get_in(&1, [:l2, :hits]),
    tags: [:cache]
  ),
  ...
]
If what you need is the aggregated stats for all levels, you can always define
your own function to emit the Telemetry events. You just need to call
MyApp.Multilevel.stats() and then you add the logic to process the results
in the way you need. On the other hand, if you are using Datadog through the
StatsD reporter, you could do the aggregation directly in Datadog.


  

    
Migrating to v2.x
    

For the v2, Nebulex introduces several breaking changes, so this guide aims to
highlight most of these changes to make easier the transition to v2. Be aware
this guide won't focus on every change, just the most significant ones that can
affect how your application code interacts with the cache. Also, it is not a
detailed guide about how to translate the current code from older versions to
v2, just pointing out the areas the new documentation should be consulted on.

  
    
  
  Configuration


This is one of the biggest changes. Version 1.x, most of the configuration
options are resolved in compile-time, which has a lot of limitations.
Since version 2.x, only few arguments are configured in compile-time when
defining a cache, e.g.: otp_app:, adapter:, and primary_storage_adapter:
(for partitioned and replicated adapters). The rest of configuration parameters
are given via config file or at startup time. For more information and examples,
see Nebulex.Cache, Nebulex.Adapters.Local, Nebulex.Adapters.Partitioned,
Nebulex.Adapters.Replicated, Nebulex.Adapters.Multilevel.

  
    
  
  Cache API


There are several changes on the Nebulex.Cache API:
	The :return option is not available anymore, so it has to be removed.
	The :version option is not available anymore, so it has to be removed.
	Callback set/3 was refactored to put/3.
	Callback set_many/2 was refactored to put_all/2.
	Callback get_many/2 was refactored to get_all/2.
	Callbacks add/3 and add!/3 were refactored to put_new/3 and
put_new!/3.
	Callback update_counter/3 was refactored to incr/3 and decr/3.
	Callback add_or_replace/3 was removed.
	Callback object_info/2 was removed, and callbacks ttl/1 and
touch/1 were added instead.


  
    
  
  Declarative annotation-based caching via decorators


	Module Nebulex.Caching.Decorators was refactored to Nebulex.Caching –
Keep in mind that since v1.2.x the caching decorators were included instead
of the previous macros or DSL (this applies for version lower than v1.1.x).
	Decorator cache/3 was refactored to cacheable/3.
	Decorator evict/3 was refactored to cache_evict/3.
	Decorator update/3 was refactored to cache_put/3.
	Improved the :match option to return not only a boolean but return a
specific value to be cached (term -> boolean | {true, term}) – If true
the code-block evaluation result is cached as it is (the default). If
{true, value} is returned, then the value is what is cached.


  
    
  
  Hooks


Since v2.x, pre/post hooks are deprecated and won't be longer supported by
Nebulex, at least not directly. Mainly, because the hooks feature is not a
common use-case and also it is something that can be be easily implemented
on top of the Cache at the application level. However, to keep backward
compatibility somehow, Nebulex provides decorators for implementing
pre/post hooks very easily. For that reason, it is highly recommended
to removed all pre/post hooks related code and adapt it to the new way.
See Nebulex.Hook for more information.

  
    
  
  Built-In Adapters


There have been several and significant improvements on the built-in adapters,
so it is also highly recommended to take a look at them;
Nebulex.Adapters.Local, Nebulex.Adapters.Partitioned,
Nebulex.Adapters.Replicated, and Nebulex.Adapters.Multilevel.
In case of using a distributed adapter, the module Nebulex.Adapter.HashSlot
was refactored to Nebulex.Adapter.Keyslot and the callback keyslot /2 to
hash_slot/2.

  
    
  
  Statistics


For older versions (<= 1.x), the stats were implemented via a post-hook and the
measurements were oriented for counting the number of times a cache function is
called. But what is interesting and useful to see is, for instance, the number
of writes, hits, misses, evictions, etc. Therefore, the whole stats'
functionality was refactored entirely.
	This feature is not longer using pre/post hooks. Besides, pre/post hooks
are deprecated in v2.x.
	The stats support is optional by implementing the Nebulex.Adapter.Stats
behaviour from the adapter. However, Nebulex provides a default
implementation using [Erlang counters][https://erlang.org/doc/man/counters.html]
which is supported by the local built-in adapter.
See the Telemetry guide for
more information.
	Since Nebulex 2.x on-wards, enabling stats is a matter of setting the
option :stats to true. See Nebulex.Cache for more information.


  
    
  
  Mix Tasks


	mix nebulex.gen.cache was refactored to mix nbx.gen.cache.
	mix nebulex was refactored to mix nbx.



  

    
Creating New Adapter
    

This guide will walk you through creating a custom Nebulex adapter. We will
start by creating a new project, making tests pass, and then implementing a
simple in-memory adapter. It will be roughly based on
NebulexRedisAdapter so you
can consult this repo as an example.

  
    
  
  Mix Project


Nebulex's main repo contains some very useful shared tests that we are going to
take advantage of. To do so we will need to checkout Nebulex from Github as the
version published to Hex does not contain test code. To accommodate this
workflow we will start by creating a new project.
mix new nebulex_memory_adapter

Now let's modify mix.exs so that we could fetch Nebulex repository.
defmodule NebulexMemoryAdapter.MixProject do
  use Mix.Project

  @nbx_vsn "2.5.2"
  @version "0.1.0"

  def project do
    [
      app: :nebulex_memory_adapter,
      version: @version,
      elixir: "~> 1.13",
      elixirc_paths: elixirc_paths(Mix.env()),
      aliases: aliases(),
      deps: deps(),
    ]
  end

  # Run "mix help compile.app" to learn about applications.
  def application do
    [
      extra_applications: [:logger]
    ]
  end

  defp elixirc_paths(:test), do: ["lib", "test/support"]
  defp elixirc_paths(_), do: ["lib"]

  # Run "mix help deps" to learn about dependencies.
  defp deps do
    [
      nebulex_dep(),
      {:telemetry, "~> 0.4 or ~> 1.0", optional: true}
    ]
  end

  defp nebulex_dep do
    if path = System.get_env("NEBULEX_PATH") do
      {:nebulex, "~> #{@nbx_vsn}", path: path}
    else
      {:nebulex, "~> #{@nbx_vsn}"}
    end
  end

  defp aliases do
    [
      "nbx.setup": [
        "cmd rm -rf nebulex",
        "cmd git clone --depth 1 --branch v#{@nbx_vsn} https://github.com/cabol/nebulex"
      ]
    ]
  end
end
As you can see here we define a mix nbx.setup task that will fetch a Nebulex
version to a folder specified in NEBULEX_PATH environmental variable. Let's
run it.
export NEBULEX_PATH=nebulex
mix nbx.setup

Now is the good time to fetch other dependencies
mix deps.get


  
    
  
  Tests


Before we start implementing our custom adapter, let's make our tests up and
running.
We start by defining a cache that uses our adapter in
test/support/test_cache.ex
defmodule NebulexMemoryAdapter.TestCache do
  use Nebulex.Cache,
    otp_app: :nebulex_memory_adapter,
    adapter: NebulexMemoryAdapter
end
We won't be writing tests ourselves. Instead, we will use shared tests from the
Nebulex parent repo. To do so, we will create a helper module in
test/shared/cache_test.exs that will use test suites for behaviour we are
going to implement. The minimal set of behaviours is Entry and Queryable so
we'll go with them.
defmodule NebulexMemoryAdapter.CacheTest do
  @moduledoc """
  Shared Tests
  """

  defmacro __using__(_opts) do
    quote do
      use Nebulex.Cache.EntryTest
      use Nebulex.Cache.QueryableTest
    end
  end
end
Let's now edit test/nebulex_memory_adapter_test.exs and make it run those
shared tests by calling use NebulexMemoryAdapter.CacheTest. We also need to
define a setup that will start our cache process and put the cache and name
keys into the test context.
defmodule NebulexMemoryAdapterTest do
  use ExUnit.Case, async: true
  use NebulexMemoryAdapter.CacheTest

  alias NebulexMemoryAdapter.TestCache, as: Cache

  setup do
    {:ok, pid} = Cache.start_link()
    Cache.delete_all()
    :ok

    on_exit(fn ->
      :ok = Process.sleep(100)
      if Process.alive?(pid), do: Cache.stop(pid)
    end)

    {:ok, cache: Cache, name: Cache}
  end
end
Now it's time to grind through failing tests.
mix test
== Compilation error in file test/support/test_cache.ex ==
** (ArgumentError) expected :adapter option given to Nebulex.Cache to list Nebulex.Adapter as a behaviour
    (nebulex 2.4.2) lib/nebulex/cache/supervisor.ex:50: Nebulex.Cache.Supervisor.compile_config/1
    test/support/test_cache.ex:2: (module)

Looks like our adapter needs to Nebulex.Adapter behaviour. Luckily, it's just
2 callback that we can copy from Nebulex.Adapters.Nil
# lib/nebulex_memory_adapter.ex
defmodule NebulexMemoryAdapter do
  @behaviour Nebulex.Adapter

  @impl Nebulex.Adapter
  defmacro __before_compile__(_env), do: :ok

  @impl Nebulex.Adapter
  def init(_opts) do
    child_spec = Supervisor.child_spec({Agent, fn -> :ok end}, id: {Agent, 1})
    {:ok, child_spec, %{}}
  end
end
Another try
mix test
== Compilation error in file test/nebulex_memory_adapter_test.exs ==
** (CompileError) test/nebulex_memory_adapter_test.exs:3: module Nebulex.Cache.EntryTest is not loaded and could not be found
    (elixir 1.13.2) expanding macro: Kernel.use/1
    test/nebulex_memory_adapter_test.exs:3: NebulexMemoryAdapterTest (module)
    expanding macro: NebulexMemoryAdapter.CacheTest.__using__/1
    test/nebulex_memory_adapter_test.exs:3: NebulexMemoryAdapterTest (module)
    (elixir 1.13.2) expanding macro: Kernel.use/1
    test/nebulex_memory_adapter_test.exs:3: NebulexMemoryAdapterTest (module)

Looks like files from Nebulex parent repo aren't automatically compiled. Let's
address this in test/test_helper.exs
# Nebulex dependency path
nbx_dep_path = Mix.Project.deps_paths()[:nebulex]

for file <- File.ls!("#{nbx_dep_path}/test/support"), file != "test_cache.ex" do
  Code.require_file("#{nbx_dep_path}/test/support/" <> file, __DIR__)
end

for file <- File.ls!("#{nbx_dep_path}/test/shared/cache") do
  Code.require_file("#{nbx_dep_path}/test/shared/cache/" <> file, __DIR__)
end

for file <- File.ls!("#{nbx_dep_path}/test/shared"), file != "cache" do
  Code.require_file("#{nbx_dep_path}/test/shared/" <> file, __DIR__)
end

# Load shared tests
for file <- File.ls!("test/shared"), not File.dir?("test/shared/" <> file) do
  Code.require_file("./shared/" <> file, __DIR__)
end

ExUnit.start()
One more attempt
mix test
< ... >
 54) test put_all/2 puts the given entries using different data types at once (NebulexMemoryAdapterTest)
     test/nebulex_memory_adapter_test.exs:128
     ** (UndefinedFunctionError) function NebulexMemoryAdapter.TestCache.delete_all/0 is undefined or private. Did you mean:

           * delete/1
           * delete/2

     stacktrace:
       (nebulex_memory_adapter 0.1.0) NebulexMemoryAdapter.TestCache.delete_all()
       test/nebulex_memory_adapter_test.exs:9: NebulexMemoryAdapterTest.__ex_unit_setup_0/1
       test/nebulex_memory_adapter_test.exs:1: NebulexMemoryAdapterTest.__ex_unit__/2



Finished in 0.2 seconds (0.2s async, 0.00s sync)
54 tests, 54 failures


  
    
  
  Implementation


Now that we have our failing tests we can write some implementation. We'll start
by making delete_all/0 work as it is called in the setup.
defmodule NebulexMemoryAdapter do
  @behaviour Nebulex.Adapter
  @behaviour Nebulex.Adapter.Queryable

  @impl Nebulex.Adapter
  defmacro __before_compile__(_env), do: :ok

  @impl Nebulex.Adapter
  def init(_opts) do
    child_spec = Supervisor.child_spec({Agent, fn -> %{} end}, id: {Agent, 1})
    {:ok, child_spec, %{}}
  end

  @impl Nebulex.Adapter.Queryable
  def execute(adapter_meta, :delete_all, query, opts) do
    deleted = Agent.get(adapter_meta.pid, &map_size/1)
    Agent.update(adapter_meta.pid, fn _state -> %{} end)

    deleted
  end
end
Did we make any progress?
mix test
< ... >

 44) test decr/3 decrements a counter by the given amount with default (NebulexMemoryAdapterTest)
     test/nebulex_memory_adapter_test.exs:355
     ** (UndefinedFunctionError) function NebulexMemoryAdapter.update_counter/6 is undefined or private
     stacktrace:
       (nebulex_memory_adapter 0.1.0) NebulexMemoryAdapter.update_counter(%{cache: NebulexMemoryAdapter.TestCache, pid: #PID<0.549.0>}, :counter1, -1, :infinity, 10, [default: 10])
       test/nebulex_memory_adapter_test.exs:356: (test)



Finished in 5.7 seconds (5.7s async, 0.00s sync)
54 tests, 44 failures

We certainly did! From here you can continue to implement necessary callbacks
one-by-one or define them all in bulk. For posterity, we put a complete
NebulexMemoryAdapter module here that passes all tests.
defmodule NebulexMemoryAdapter do
  @behaviour Nebulex.Adapter
  @behaviour Nebulex.Adapter.Entry
  @behaviour Nebulex.Adapter.Queryable

  @impl Nebulex.Adapter
  defmacro __before_compile__(_env), do: :ok

  @impl Nebulex.Adapter
  def init(_opts) do
    child_spec = Supervisor.child_spec({Agent, fn -> %{} end}, id: {Agent, 1})
    {:ok, child_spec, %{}}
  end

  @impl Nebulex.Adapter.Entry
  def get(adapter_meta, key, _opts) do
    Agent.get(adapter_meta.pid, &Map.get(&1, key))
  end

  @impl Nebulex.Adapter.Entry
  def get_all(adapter_meta, keys, _opts) do
    Agent.get(adapter_meta.pid, &Map.take(&1, keys))
  end

  @impl Nebulex.Adapter.Entry
  def put(adapter_meta, key, value, ttl, :put_new, opts) do
    if get(adapter_meta, key, []) do
      false
    else
      put(adapter_meta, key, value, ttl, :put, opts)
      true
    end
  end

  def put(adapter_meta, key, value, ttl, :replace, opts) do
    if get(adapter_meta, key, []) do
      put(adapter_meta, key, value, ttl, :put, opts)
      true
    else
      false
    end
  end

  def put(adapter_meta, key, value, _ttl, _on_write, _opts) do
    Agent.update(adapter_meta.pid, &Map.put(&1, key, value))
    true
  end

  @impl Nebulex.Adapter.Entry
  def put_all(adapter_meta, entries, ttl, :put_new, opts) do
    if get_all(adapter_meta, Map.keys(entries), []) == %{} do
      put_all(adapter_meta, entries, ttl, :put, opts)
      true
    else
      false
    end
  end

  def put_all(adapter_meta, entries, _ttl, _on_write, _opts) do
    entries = Map.new(entries)
    Agent.update(adapter_meta.pid, &Map.merge(&1, entries))
    true
  end

  @impl Nebulex.Adapter.Entry
  def delete(adapter_meta, key, _opts) do
    Agent.update(adapter_meta.pid, &Map.delete(&1, key))
  end

  @impl Nebulex.Adapter.Entry
  def take(adapter_meta, key, _opts) do
    value = get(adapter_meta, key, [])
    delete(adapter_meta, key, [])
    value
  end

  @impl Nebulex.Adapter.Entry
  def update_counter(adapter_meta, key, amount, _ttl, default, _opts) do
    Agent.update(adapter_meta.pid, fn state ->
      Map.update(state, key, default + amount, fn v -> v + amount end)
    end)

    get(adapter_meta, key, [])
  end

  @impl Nebulex.Adapter.Entry
  def has_key?(adapter_meta, key) do
    Agent.get(adapter_meta.pid, &Map.has_key?(&1, key))
  end

  @impl Nebulex.Adapter.Entry
  def ttl(_adapter_meta, _key) do
    nil
  end

  @impl Nebulex.Adapter.Entry
  def expire(_adapter_meta, _key, _ttl) do
    true
  end

  @impl Nebulex.Adapter.Entry
  def touch(_adapter_meta, _key) do
    true
  end

  @impl Nebulex.Adapter.Queryable
  def execute(adapter_meta, :delete_all, _query, _opts) do
    deleted = execute(adapter_meta, :count_all, nil, [])
    Agent.update(adapter_meta.pid, fn _state -> %{} end)

    deleted
  end

  def execute(adapter_meta, :count_all, _query, _opts) do
    Agent.get(adapter_meta.pid, &map_size/1)
  end

  def execute(adapter_meta, :all, _query, _opts) do
    Agent.get(adapter_meta.pid, &Map.values/1)
  end

  @impl Nebulex.Adapter.Queryable
  def stream(_adapter_meta, :invalid_query, _opts) do
    raise Nebulex.QueryError, message: "foo", query: :invalid_query
  end

  def stream(adapter_meta, _query, opts) do
    fun =
      case Keyword.get(opts, :return) do
        :value ->
          &Map.values/1

        {:key, :value} ->
          &Map.to_list/1

        _ ->
          &Map.keys/1
      end

    Agent.get(adapter_meta.pid, fun)
  end
end
Of course, this isn't a useful adapter in any sense but it should be enough to
get you started with your own.

Nebulex is split into 2 main components:
	Nebulex.Cache - caches are wrappers around the in-memory data store.
Via the cache, we can put, get, update, delete and query existing entries.
A cache needs an adapter to communicate to the in-memory data store.

	Nebulex.Caching - Declarative annotation-based caching via
Nebulex.Caching.Decorators. Decorators provide n elegant way of
annotating functions to be cached or evicted. Caching decorators also
enable the usage and/or implementation of cache usage patterns like
Read-through, Write-through, Cache-as-SoR, etc.
See Cache Usage Patterns Guide.


In the following sections, we will provide an overview of those components and
how they interact with each other. Feel free to access their respective module
documentation for more specific examples, options and configuration.
If you want to quickly check a sample application using Nebulex, please check
the getting started guide.

  
    
  
  Caches


Nebulex.Cache is the wrapper around the Cache. We can define a
cache as follows:
defmodule MyApp.MyCache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Local
end
Where the configuration for the Cache must be in your application
environment, usually defined in your config/config.exs:
config :my_app, MyApp.MyCache,
  gc_interval: 3_600_000, #=> 1 hr
  backend: :shards,
  partitions: 2
Each cache in Nebulex defines a start_link/1 function that needs to be
invoked before using the cache. In general, this function is not called
directly, but used as part of your application supervision tree.
If your application was generated with a supervisor (by passing --sup
to mix new) you will have a lib/my_app/application.ex file containing
the application start callback that defines and starts your supervisor.
You just need to edit the start/2 function to start the cache as a
supervisor on your application's supervisor:
def start(_type, _args) do
  children = [
    {MyApp.Cache, []}
  ]

  opts = [strategy: :one_for_one, name: MyApp.Supervisor]
  Supervisor.start_link(children, opts)
end
Otherwise, you can start and stop the cache directly at any time by calling
MyApp.Cache.start_link/1 and MyApp.Cache.stop/1.

  
    
  
  Declarative annotation-based caching


See Nebulex.Caching.

Specifies the minimal API required from adapters.

Nebulex.Adapter.Entry behaviour
    



      
Specifies the entry API required from adapters.
This behaviour specifies all read/write key-based functions,
the ones applied to a specific cache entry.

This behaviour provides a callback to compute the hash slot for a specific
key based on the number of slots (partitions, nodes, ...).
The purpose of this module is to allow users to implement a custom
hash-slot function to distribute the keys. It can be used to select
the node/slot where a specific key is supposed to be.
It is highly recommended to use a Consistent Hashing algorithm.


  
    
  
  Example


defmodule MyApp.Keyslot do
  use Nebulex.Adapter.Keyslot

  @impl true
  def hash_slot(key, range) do
    key
    |> :erlang.phash2()
    |> :jchash.compute(range)
  end
end
This example uses Jumping Consistent Hash.

Nebulex.Adapter.Persistence behaviour
    



      
Specifies the adapter persistence API.

  
    
  
  Default implementation


This module provides a default implementation that uses File and Stream
under-the-hood. For dumping a cache to a file, the entries are streamed from
the cache and written in chunks (one chunk per line), and each chunk contains
N number of entries. For loading the entries from a file, the file is read
and streamed line-by-line, so that the entries collected on each line are
inserted in streaming fashion as well.
The default implementation accepts the following options only for dump
operation (there are not options for load):
	entries_per_line - The number of entries to be written per line in the
file. Defaults to 10.

	compression - The compression level. The values are the same as
:erlang.term_to_binary /2. Defaults to 6.


See Nebulex.Cache.dump/2 and Nebulex.Cache.load/2 for more
information.

Specifies the query API required from adapters.

  
    
  
  Query values


There are two types of query values. The ones shared and implemented
by all adapters and the ones that are adapter specific.

  
    
  
  Common queries


The following query values are shared and/or supported for all adapters:
	nil - Matches all cached entries.


  
    
  
  Adapter-specific queries


The query value depends entirely on the adapter implementation; it could
any term. Therefore, it is highly recommended to see adapters' documentation
for more information about building queries. For example, the built-in
Nebulex.Adapters.Local adapter uses :ets.match_spec() for queries,
as well as other pre-defined ones like :unexpired and :expired.

Specifies the stats API required from adapters.
Each adapter is responsible for providing support for stats by implementing
this behaviour. However, this module brings with a default implementation
using [Erlang counters][https://erlang.org/doc/man/counters.html], with all
callbacks overridable, which is supported by the built-in adapters.
See Nebulex.Adapters.Local for more information about how this can be used
from the adapter, and also Nebulex Telemetry Guide to learn
how to use the Cache with Telemetry.

      


      
        Summary


  
    Callbacks
  


    
      
        
  
    
    Nebulex.Adapter.Transaction - Nebulex v2.6.5
    
    

    


  
  

    
Nebulex.Adapter.Transaction behaviour
    



      
Specifies the adapter transactions API.

  
    
  
  Default implementation


This module also provides a default implementation which uses the Erlang
library :global.
This implementation accepts the following options:
	:keys - The list of the keys that will be locked. Since the lock id is
generated based on the key, if this option is not set, a fixed/constant
lock id is used to perform the transaction, then all further transactions
(without this option set) are serialized and the performance is affected
significantly. For that reason it is recommended to pass the list of keys
involved in the transaction.

	:nodes - The list of the nodes where the lock will be set, or on
all nodes if none are specified.

	:retries - If the key has been locked by other process already, and
:retries is not equal to 0, the process sleeps for a while and tries
to execute the action later. When :retries attempts have been made,
an exception is raised. If :retries is :infinity (the default),
the function will eventually be executed (unless the lock is never
released).


Let's see an example:
MyCache.transaction fn ->
  counter = MyCache.get(:counter)
  MyCache.set(:counter, counter + 1)
end
Locking only the involved key (recommended):
MyCache.transaction [keys: [:counter]], fn ->
  counter = MyCache.get(:counter)
  MyCache.set(:counter, counter + 1)
end

MyCache.transaction [keys: [:alice, :bob]], fn ->
  alice = MyCache.get(:alice)
  bob = MyCache.get(:bob)
  MyCache.set(:alice, %{alice | balance: alice.balance + 100})
  MyCache.set(:bob, %{bob | balance: bob.balance + 100})
end

Adapter module for Local Generational Cache; inspired by
epocxy.
Generational caching using an ets table (or multiple ones when used with
:shards) for each generation of cached data. Accesses hit the newer
generation first, and migrate from the older generation to the newer
generation when retrieved from the stale table. When a new generation
is started, the oldest one is deleted. This is a form of mass garbage
collection which avoids using timers and expiration of individual
cached elements.
This implementation of generation cache uses only two generations
(which is more than enough) also referred like the newer and
the older.

  
    
  
  Overall features


	Configurable backend (ets or :shards).
	Expiration – A status based on TTL (Time To Live) option. To maintain
cache performance, expired entries may not be immediately removed or
evicted, they are expired or evicted on-demand, when the key is read.
	Eviction – Generational Garbage Collection.
	Sharding – For intensive workloads, the Cache may also be partitioned
(by using :shards backend and specifying the :partitions option).
	Support for transactions via Erlang global name registration facility.
	Support for stats.


  
    
  
  Options


This adapter supports the following options and all of them can be given via
the cache configuration:
	:backend - Defines the backend or storage to be used for the adapter.
Supported backends are: :ets and :shards. Defaults to :ets.

	:read_concurrency - (boolean) Since this adapter uses ETS tables
internally, this option is used when a new table is created; see
:ets.new/2. Defaults to true.

	:write_concurrency - (boolean) Since this adapter uses ETS tables
internally, this option is used when a new table is created; see
:ets.new/2. Defaults to true.

	:compressed - (boolean) This option is used when a new ETS table is
created and it defines whether or not it includes X as an option; see
:ets.new/2. Defaults to false.

	:backend_type - This option defines the type of ETS to be used
(Defaults to :set). However, it is highly recommended to keep the
default value, since there are commands not supported (unexpected
exception may be raised) for types like :bag or : duplicate_bag.
Please see the ETS docs
for more information.

	:partitions - If it is set, an integer > 0 is expected, otherwise,
it defaults to System.schedulers_online(). This option is only
available for :shards backend.

	:gc_interval - If it is set, an integer > 0 is expected defining the
interval time in milliseconds to garbage collection to run, delete the
oldest generation and create a new one. If this option is not set,
garbage collection is never executed, so new generations must be
created explicitly, e.g.: MyCache.new_generation(opts).

	:max_size - If it is set, an integer > 0 is expected defining the
max number of cached entries (cache limit). If it is not set (nil),
the check to release memory is not performed (the default).

	:allocated_memory - If it is set, an integer > 0 is expected defining
the max size in bytes allocated for a cache generation. When this option
is set and the configured value is reached, a new cache generation is
created so the oldest is deleted and force releasing memory space.
If it is not set (nil), the cleanup check to release memory is
not performed (the default).

	:gc_cleanup_min_timeout - An integer > 0 defining the min timeout in
milliseconds for triggering the next cleanup and memory check. This will
be the timeout to use when either the max size or max allocated memory
is reached. Defaults to 10_000 (10 seconds).

	:gc_cleanup_max_timeout - An integer > 0 defining the max timeout in
milliseconds for triggering the next cleanup and memory check. This is
the timeout used when the cache starts and there are few entries or the
consumed memory is near to 0. Defaults to 600_000 (10 minutes).

	:gc_flush_delay - If it is set, an integer > 0 is expected defining the
delay in milliseconds before objects from the oldest generation are
flushed. Defaults to 10_000 (10 seconds).



  
    
  
  Usage


Nebulex.Cache is the wrapper around the cache. We can define a
local cache as follows:
defmodule MyApp.LocalCache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Local
end
Where the configuration for the cache must be in your application
environment, usually defined in your config/config.exs:
config :my_app, MyApp.LocalCache,
  gc_interval: :timer.hours(12),
  max_size: 1_000_000,
  allocated_memory: 2_000_000_000,
  gc_cleanup_min_timeout: :timer.seconds(10),
  gc_cleanup_max_timeout: :timer.minutes(10)
For intensive workloads, the Cache may also be partitioned using :shards
as cache backend (backend: :shards) and configuring the desired number of
partitions via the :partitions option. Defaults to
System.schedulers_online().
config :my_app, MyApp.LocalCache,
  gc_interval: :timer.hours(12),
  max_size: 1_000_000,
  allocated_memory: 2_000_000_000,
  gc_cleanup_min_timeout: :timer.seconds(10),
  gc_cleanup_max_timeout: :timer.minutes(10),
  backend: :shards,
  partitions: System.schedulers_online() * 2
If your application was generated with a supervisor (by passing --sup
to mix new) you will have a lib/my_app/application.ex file containing
the application start callback that defines and starts your supervisor.
You just need to edit the start/2 function to start the cache as a
supervisor on your application's supervisor:
def start(_type, _args) do
  children = [
    {MyApp.LocalCache, []},
    ...
  ]
See Nebulex.Cache for more information.

  
    
  
  Eviction configuration


This section is to understand a bit better how the different configuration
options work and have an idea what values to set; especially if it is the
first time using Nebulex.

  
    
  
  :ttl option


The :ttl option that is used to set the expiration time for a key, it
doesn't work as eviction mechanism, since the local adapter implements a
generational cache, the options that control the eviction process are:
:gc_interval, :gc_cleanup_min_timeout, :gc_cleanup_max_timeout,
:max_size and :allocated_memory. The :ttl is evaluated on-demand
when a key is retrieved, and at that moment if it s expired, then remove
it from the cache, hence, it can not be used as eviction method, it is
more for keep the integrity and consistency in the cache. For this reason,
it is highly recommended to configure always the eviction options mentioned
before.

  
    
  
  Caveats when using :ttl option:


	When using the :ttl option, ensure it is less than :gc_interval,
otherwise, there may be a situation where the key is evicted and the
:ttl hasn't happened yet (maybe because the garbage collector ran
before the key had been fetched).
	Assuming you have :gc_interval set to 2 hrs, then you put a new key
with :ttl set to 1 hr, and 1 minute later the GC runs, that key will
be moved to the older generation so it can be yet retrieved. On the other
hand, if the key is never fetched till the next GC cycle (causing moving
it to the newer generation), since the key is already in the oldest
generation it will be evicted from the cache so it won't be retrievable
anymore.


  
    
  
  Garbage collection or eviction options


This adapter implements a generational cache, which means its main eviction
mechanism is pushing a new cache generation and remove the oldest one. In
this way, we ensure only the most frequently used keys are always available
in the newer generation and the the least frequently used are evicted when
the garbage collector runs, and the garbage collector is triggered upon
these conditions:
	When the time interval defined by :gc_interval is completed.
This makes the garbage-collector process to run creating a new
generation and forcing to delete the oldest one.
	When the "cleanup" timeout expires, and then the limits :max_size
and :allocated_memory are checked, if one of those is reached,
then the garbage collector runs (a new generation is created and
the oldest one is deleted). The cleanup timeout is controlled by
:gc_cleanup_min_timeout and :gc_cleanup_max_timeout, it works
with an inverse linear backoff, which means the timeout is inverse
proportional to the memory growth; the bigger the cache size is,
the shorter the cleanup timeout will be.


  
    
  
  First-time configuration


For configuring the cache with accurate and/or good values it is important
to know several things in advance, like for example the size of an entry
in average so we can calculate a good value for max size and/or allocated
memory, how intensive will be the load in terms of reads and writes, etc.
The problem is most of these aspects are unknown when it is a new app or
we are using the cache for the first time. Therefore, the following
recommendations will help you to configure the cache for the first time:
	When configuring the :gc_interval, think about how that often the
least frequently used entries should be evicted, or what is the desired
retention period for the cached entries. For example, if :gc_interval
is set to 1 hr, it means you will keep in cache only those entries that
are retrieved periodically within a 2 hr period; gc_interval * 2,
being 2 the number of generations. Longer than that, the GC will
ensure is always evicted (the oldest generation is always deleted).
If it is the first time using Nebulex, perhaps you can start with
gc_interval: :timer.hours(12) (12 hrs), so the max retention
period for the keys will be 1 day; but ensure you also set either the
:max_size or :allocated_memory.
	It is highly recommended to set either :max_size or :allocated_memory
to ensure the oldest generation is deleted (least frequently used keys
are evicted) when one of these limits is reached and also to avoid
running out of memory. For example, for the :allocated_memory we can
set 25% of the total memory, and for the :max_size something between
100_000 and 1_000_000.
	For :gc_cleanup_min_timeout we can set 10_000, which means when the
cache is reaching the size or memory limit, the polling period for the
cleanup process will be 10 seconds. And for :gc_cleanup_max_timeout
we can set 600_000, which means when the cache is almost empty the
polling period will be close to 10 minutes.


  
    
  
  Stats


This adapter does support stats by using the default implementation
provided by Nebulex.Adapter.Stats. The adapter also uses the
Nebulex.Telemetry.StatsHandler to aggregate the stats and keep
them updated. Therefore, it requires the Telemetry events are emitted
by the adapter (the :telemetry option should not be set to false
so the Telemetry events can be dispatched), otherwise, stats won't
work properly.

  
    
  
  Queryable API


Since this adapter is implemented on top of ETS tables, the query must be
a valid match spec given by :ets.match_spec(). However, there are some
predefined and/or shorthand queries you can use. See the section
"Predefined queries" below for for information.
Internally, an entry is represented by the tuple
{:entry, key, value, touched, ttl}, which means the match pattern within
the :ets.match_spec() must be something like:
{:entry, :"$1", :"$2", :"$3", :"$4"}.
In order to make query building easier, you can use Ex2ms library.

  
    
  
  Predefined queries


	nil - All keys are returned.

	:unexpired -  All unexpired keys/entries.

	:expired -  All expired keys/entries.

	{:in, [term]} - Only the keys in the given key list ([term])
are returned. This predefined query is only supported for
Nebulex.Cache.delete_all/2. This is the recommended
way of doing bulk delete of keys.



  
    
  
  Examples


# built-in queries
MyCache.all()
MyCache.all(:unexpired)
MyCache.all(:expired)
MyCache.all({:in, ["foo", "bar"]})

# using a custom match spec (all values > 10)
spec = [{{:_, :"$1", :"$2", :_, :_}, [{:>, :"$2", 10}], [{{:"$1", :"$2"}}]}]
MyCache.all(spec)

# using Ex2ms
import Ex2ms

spec =
  fun do
    {_, key, value, _, _} when value > 10 -> {key, value}
  end

MyCache.all(spec)
The :return option applies only for built-in queries, such as:
nil | :unexpired | :expired, if you are using a custom :ets.match_spec(),
the return value depends on it.
The same applies to the stream function.

  
    
  
  Extended API (convenience functions)


This adapter provides some additional convenience functions to the
Nebulex.Cache API.
Creating new generations:
MyCache.new_generation()
MyCache.new_generation(reset_timer: false)
Retrieving the current generations:
MyCache.generations()
Retrieving the newer generation:
MyCache.newer_generation()

Generational garbage collection process.
The generational garbage collector manage the heap as several sub-heaps,
known as generations, based on age of the objects. An object is allocated
in the youngest generation, sometimes called the nursery, and is promoted
to an older generation if its lifetime exceeds the threshold of its current
generation (defined by option :gc_interval). Every time the GC runs
(triggered by :gc_interval timeout), a new cache generation is created
and the oldest one is deleted.
The deletion of the oldest generation happens in two steps. First, the
underlying ets table is flushed to release space and only marked for deletion
as there may still be processes referencing it. The actual deletion of the
ets table happens at next GC run.
However, flushing is a blocking operation, once started, processes wanting
to access the table will need to wait until it finishes. To circumvent this,
flushing can be delayed by configuring :gc_flush_delay to allow time for
these processes to finish their work without being accidentally blocked.
The only way to create new generations is through this module (this server
is the metadata owner) calling new/2 function. When a Cache is created,
a generational garbage collector is attached to it automatically,
therefore, this server MUST NOT be started directly.

  
    
  
  Options


These options are configured through the Nebulex.Adapters.Local adapter:
	:gc_interval - If it is set, an integer > 0 is expected defining the
interval time in milliseconds to garbage collection to run, delete the
oldest generation and create a new one. If this option is not set,
garbage collection is never executed, so new generations must be
created explicitly, e.g.: MyCache.new_generation(opts).

	:max_size - If it is set, an integer > 0 is expected defining the
max number of cached entries (cache limit). If it is not set (nil),
the check to release memory is not performed (the default).

	:allocated_memory - If it is set, an integer > 0 is expected defining
the max size in bytes allocated for a cache generation. When this option
is set and the configured value is reached, a new cache generation is
created so the oldest is deleted and force releasing memory space.
If it is not set (nil), the cleanup check to release memory is
not performed (the default).

	:gc_cleanup_min_timeout - An integer > 0 defining the min timeout in
milliseconds for triggering the next cleanup and memory check. This will
be the timeout to use when either the max size or max allocated memory
is reached. Defaults to 10_000 (10 seconds).

	:gc_cleanup_max_timeout - An integer > 0 defining the max timeout in
milliseconds for triggering the next cleanup and memory check. This is
the timeout used when the cache starts and there are few entries or the
consumed memory is near to 0. Defaults to 600_000 (10 minutes).

	:gc_flush_delay - If it is set, an integer > 0 is expected defining the
delay in milliseconds before objects from the oldest generation are
flushed. Defaults to 10_000 (10 seconds).

Adapter module for Multi-level Cache.
This is just a simple layer on top of local or distributed cache
implementations that enables to have a cache hierarchy by levels.
Multi-level caches generally operate by checking the fastest,
level 1 (L1) cache first; if it hits, the adapter proceeds at
high speed. If that first cache misses, the next fastest cache
(level 2, L2) is checked, and so on, before accessing external
memory (that can be handled by a cacheable decorator).
For write functions, the "Write Through" policy is applied by default;
this policy ensures that the data is stored safely as it is written
throughout the hierarchy. However, it is possible to force the write
operation in a specific level (although it is not recommended) via
level option, where the value is a positive integer greater than 0.
We can define a multi-level cache as follows:
defmodule MyApp.Multilevel do
  use Nebulex.Cache,
    otp_app: :nebulex,
    adapter: Nebulex.Adapters.Multilevel

  defmodule L1 do
    use Nebulex.Cache,
      otp_app: :nebulex,
      adapter: Nebulex.Adapters.Local
  end

  defmodule L2 do
    use Nebulex.Cache,
      otp_app: :nebulex,
      adapter: Nebulex.Adapters.Partitioned
  end
end
Where the configuration for the cache and its levels must be in your
application environment, usually defined in your config/config.exs:
config :my_app, MyApp.Multilevel,
  model: :inclusive,
  levels: [
    {
      MyApp.Multilevel.L1,
      gc_interval: :timer.hours(12),
      backend: :shards
    },
    {
      MyApp.Multilevel.L2,
      primary: [
        gc_interval: :timer.hours(12),
        backend: :shards
      ]
    }
  ]
If your application was generated with a supervisor (by passing --sup
to mix new) you will have a lib/my_app/application.ex file containing
the application start callback that defines and starts your supervisor.
You just need to edit the start/2 function to start the cache as a
supervisor on your application's supervisor:
def start(_type, _args) do
  children = [
    {MyApp.Multilevel, []},
    ...
  ]
See Nebulex.Cache for more information.

  
    
  
  Options


This adapter supports the following options and all of them can be given via
the cache configuration:
	:levels - This option is to define the levels, a list of tuples
{cache_level :: Nebulex.Cache.t(), opts :: Keyword.t()}, where
the first element is the module that defines the cache for that
level, and the second one is the options that will be passed to
that level in the start/link/1 (which depends on the adapter
this level is using). The order in which the levels are defined
is the same the multi-level cache will use. For example, the first
cache in the list will be the L1 cache (level 1) and so on;
the Nth element will be the LN cache. This option is mandatory,
if it is not set or empty, an exception will be raised.

	:model - Specifies the cache model: :inclusive or :exclusive;
defaults to :inclusive. In an inclusive cache, the same data can be
present in all caches/levels. In an exclusive cache, data can be present
in only one cache/level and a key cannot be found in the rest of caches
at the same time. This option applies to the get callback only; if the
cache :model is :inclusive, when the key is found in a level N,
that entry is duplicated backwards (to all previous levels: 1..N-1).
However, when the mode is set to :inclusive, the get_all operation
is translated into multiple get calls underneath (which may be a
significant performance penalty) since is required to replicate the
entries properly with their current TTLs. It is possible to skip the
replication when calling get_all using the option :replicate.

	:replicate - This option applies only to the get_all callback.
Determines whether the entries should be replicated to the backward
levels or not. Defaults to true.



  
    
  
  Shared options


Almost all of the cache functions outlined in Nebulex.Cache module
accept the following options:
	:level - It may be an integer greater than 0 that specifies the cache
level where the operation will take place. By default, the evaluation
is performed throughout the whole cache hierarchy (all levels).


  
    
  
  Telemetry events


This adapter emits all recommended Telemetry events, and documented
in Nebulex.Cache module (see "Adapter-specific events" section).
Since the multi-level adapter is a layer/wrapper on top of other existing
adapters, each cache level may Telemetry emit events independently.
For example, for the cache defined before MyApp.Multilevel, the next
events will be emitted for the main multi-level cache:
	[:my_app, :multilevel, :command, :start]
	[:my_app, :multilevel, :command, :stop]
	[:my_app, :multilevel, :command, :exception]

For the L1 (configured with the local adapter):
	[:my_app, :multilevel, :l1, :command, :start]
	[:my_app, :multilevel, :l1, :command, :stop]
	[:my_app, :multilevel, :l1, :command, :exception]

For the L2 (configured with the partitioned adapter):
	[:my_app, :multilevel, :l2, :command, :start]
	[:my_app, :multilevel, :l2, :primary, :command, :start]
	[:my_app, :multilevel, :l2, :command, :stop]
	[:my_app, :multilevel, :l2, :primary, :command, :stop]
	[:my_app, :multilevel, :l2, :command, :exception]
	[:my_app, :multilevel, :l2, :primary, :command, :exception]

See also the Telemetry guide
for more information and examples.

  
    
  
  Stats


Since the multi-level adapter works as a wrapper for the configured cache
levels, the support for stats depends on the underlying levels. Also, the
measurements are consolidated per level, they are not aggregated. For example,
if we enable the stats for the multi-level cache defined previously and run:
MyApp.Multilevel.stats()
The returned stats will look like:
%Nebulex.Stats{
  measurements: %{
    l1: %{evictions: 0, expirations: 0, hits: 0, misses: 0, writes: 0},
    l2: %{evictions: 0, expirations: 0, hits: 0, misses: 0, writes: 0}
  },
  metadata: %{
    l1: %{
      cache: NMyApp.Multilevel.L1,
      started_at: ~U[2021-01-10 13:06:04.075084Z]
    },
    l2: %{
      cache: MyApp.Multilevel.L2.Primary,
      started_at: ~U[2021-01-10 13:06:04.089888Z]
    },
    cache: MyApp.Multilevel,
    started_at: ~U[2021-01-10 13:06:04.066750Z]
  }
}
IMPORTANT: Those cache levels with stats disabled won't be included
into the returned stats (they are skipped). If a cache level is using
an adapter that does not support stats, you may get unexpected errors.
Therefore, and as overall recommendation, check out the documentation
for adapters used by the underlying cache levels and ensure they
implement the Nebulex.Adapter.Stats behaviour.

  
    
  
  Stats with Telemetry


In case you are using Telemetry metrics, you can define the metrics per
level, for example:
last_value("nebulex.cache.stats.l1.hits",
  event_name: "nebulex.cache.stats",
  measurement: &get_in(&1, [:l1, :hits]),
  tags: [:cache]
)
last_value("nebulex.cache.stats.l1.misses",
  event_name: "nebulex.cache.stats",
  measurement: &get_in(&1, [:l1, :misses]),
  tags: [:cache]
)
See the section "Instrumenting Multi-level caches" in the
Telemetry guide
for more information.


  
    
  
  Extended API


This adapter provides one additional convenience function for retrieving
the cache model for the given cache name:
MyCache.model()
MyCache.model(:cache_name)

  
    
  
  Caveats of multi-level adapter


Because this adapter reuses other existing/configured adapters, it inherits
all their limitations too. Therefore, it is highly recommended to check the
documentation of the adapters to use.

Nebulex.Adapters.Nil 
    



      
The Nil adapter is a special cache adapter that disables the cache;
it loses all the items saved on it and it returns nil for all the read
and true for all save operations. This adapter is mostly useful for tests.

  
    
  
  Example


Suppose you have an application using Ecto for database access and Nebulex
for caching. Then, you have defined a cache and a repo within it. Since you
are using a database, there might be some cases you may want to disable the
cache to avoid issues when running the test, for example, in some test cases,
when accessing the database you expect no data at all, but you could retrieve
the data from cache anyway because maybe it was cached in a previous test.
Therefore, you have to delete all entries from the cache before to run each
test to make sure the cache is always empty. This is where the Nil adapter
comes in, instead of adding code to flush the cache before each test, you
could define a test cache using the Nil adapter for the tests.
One one hand, you have defined the cache in your application within
lib/my_app/cache.ex:
defmodule MyApp.Cache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Local
end
And on the other hand, in the tests you have defined the test cache within
test/support/test_cache.ex:
defmodule MyApp.TestCache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Nil
end
Now, we have to tell the app what cache to use depending on the environment,
for tests we want MyApp.TestCache, otherwise it is always MyApp.Cache.
We can do this very easy by introducing a new config parameter to decide
what cache module to use. For tests you can define the config
config/test.exs:
config :my_app,
  nebulex_cache: MyApp.TestCache,
  ...
The final piece is to read the config parameter and start the cache properly.
Within lib/my_app/application.ex you could have:
def start(_type, _args) do
  children = [
    {Application.get_env(:my_app, :nebulex_cache, MyApp.Cache), []},
  ]

  ...
As you can see, by default MyApp.Cache is always used, unless the
:nebulex_cache option points to a different module, which will be
when tests are executed (:test env).

Built-in adapter for partitioned cache topology.

  
    
  
  Overall features


	Partitioned cache topology (Sharding Distribution Model).
	Configurable primary storage adapter.
	Configurable Keyslot to distributed the keys across the cluster members.
	Support for transactions via Erlang global name registration facility.
	Stats support rely on the primary storage adapter.


  
    
  
  Partitioned Cache Topology


There are several key points to consider about a partitioned cache:
	Partitioned: The data in a distributed cache is spread out over
all the servers in such a way that no two servers are responsible for
the same piece of cached data. This means that the size of the cache
and the processing power associated with the management of the cache
can grow linearly with the size of the cluster. Also, it means that
operations against data in the cache can be accomplished with a
"single hop," in other words, involving at most one other server.

	Load-Balanced:  Since the data is spread out evenly over the
servers, the responsibility for managing the data is automatically
load-balanced across the cluster.

	Ownership: Exactly one node in the cluster is responsible for each
piece of data in the cache.

	Point-To-Point: The communication for the partitioned cache is all
point-to-point, enabling linear scalability.

	Location Transparency: Although the data is spread out across
cluster nodes, the exact same API is used to access the data, and the
same behavior is provided by each of the API methods. This is called
location transparency, which means that the developer does not have to
code based on the topology of the cache, since the API and its behavior
will be the same with a local cache, a replicated cache, or a distributed
cache.

	Failover: Failover of a distributed cache involves promoting backup
data to be primary storage. When a cluster node fails, all remaining
cluster nodes determine what data each holds in backup that the failed
cluster node had primary responsible for when it died. Those data becomes
the responsibility of whatever cluster node was the backup for the data.
However, this adapter does not provide fault-tolerance implementation,
each piece of data is kept in a single node/machine (via sharding), then,
if a node fails, the data kept by this node won't be available for the
rest of the cluster members.


Based on "Distributed Caching Essential Lessons" by Cameron Purdy
and Coherence Partitioned Cache Service.


  
    
  
  Additional implementation notes


:pg2 or :pg (>= OTP 23) is used under-the-hood by the adapter to manage
the cluster nodes. When the partitioned cache is started in a node, it creates
a group and joins it (the cache supervisor PID is joined to the group). Then,
when a function is invoked, the adapter picks a node from the group members,
and then the function is executed on that specific node. In the same way,
when a partitioned cache supervisor dies (the cache is stopped or killed for
some reason), the PID of that process is automatically removed from the PG
group; this is why it's recommended to use consistent hashing for distributing
the keys across the cluster nodes.
NOTE: pg2 will be replaced by pg in future, since the pg2 module
is deprecated as of OTP 23 and scheduled for removal in OTP 24.

This adapter depends on a local cache adapter (primary storage), it adds
a thin layer on top of it in order to distribute requests across a group
of nodes, where is supposed the local cache is running already. However,
you don't need to define any additional cache module for the primary
storage, instead, the adapter initializes it automatically (it adds the
primary storage as part of the supervision tree) based on the given
options within the primary_storage_adapter: argument.

  
    
  
  Usage


When used, the Cache expects the :otp_app and :adapter as options.
The :otp_app should point to an OTP application that has the cache
configuration. For example:
defmodule MyApp.PartitionedCache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Partitioned
end
Optionally, you can configure the desired primary storage adapter with the
option :primary_storage_adapter; defaults to Nebulex.Adapters.Local.
defmodule MyApp.PartitionedCache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Partitioned,
    primary_storage_adapter: Nebulex.Adapters.Local
end
Also, you can provide a custom keyslot function:
defmodule MyApp.PartitionedCache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Partitioned,
    primary_storage_adapter: Nebulex.Adapters.Local

  @behaviour Nebulex.Adapter.Keyslot

  @impl true
  def hash_slot(key, range) do
    key
    |> :erlang.phash2()
    |> :jchash.compute(range)
  end
end
Where the configuration for the cache must be in your application environment,
usually defined in your config/config.exs:
config :my_app, MyApp.PartitionedCache,
  keyslot: MyApp.PartitionedCache,
  primary: [
    gc_interval: 3_600_000,
    backend: :shards
  ]
If your application was generated with a supervisor (by passing --sup
to mix new) you will have a lib/my_app/application.ex file containing
the application start callback that defines and starts your supervisor.
You just need to edit the start/2 function to start the cache as a
supervisor on your application's supervisor:
def start(_type, _args) do
  children = [
    {MyApp.PartitionedCache, []},
    ...
  ]
See Nebulex.Cache for more information.

  
    
  
  Options


This adapter supports the following options and all of them can be given via
the cache configuration:
	:primary - The options that will be passed to the adapter associated
with the local primary storage. These options will depend on the local
adapter to use.

	:keyslot - Defines the module implementing Nebulex.Adapter.Keyslot
behaviour.

	:task_supervisor_opts - Start-time options passed to
Task.Supervisor.start_link/1 when the adapter is initialized.

	:join_timeout - Interval time in milliseconds for joining the
running partitioned cache to the cluster. This is to ensure it is
always joined. Defaults to :timer.seconds(180).



  
    
  
  Shared options


Almost all of the cache functions outlined in Nebulex.Cache module
accept the following options:
	:timeout - The time-out value in milliseconds for the command that
will be executed. If the timeout is exceeded, then the current process
will exit. For executing a command on remote nodes, this adapter uses
Task.await/2 internally for receiving the result, so this option tells
how much time the adapter should wait for it. If the timeout is exceeded,
the task is shut down but the current process doesn't exit, only the
result associated with that task is skipped in the reduce phase.


  
    
  
  Telemetry events


This adapter emits all recommended Telemetry events, and documented
in Nebulex.Cache module (see "Adapter-specific events" section).
Since the partitioned adapter depends on the configured primary storage
adapter (local cache adapter), this one may also emit Telemetry events.
Therefore, there will be events emitted by the partitioned adapter as well
as the primary storage adapter. For example, for the cache defined before
MyApp.PartitionedCache, these would be the emitted events:
	[:my_app, :partitioned_cache, :command, :start]
	[:my_app, :partitioned_cache, :primary, :command, :start]
	[:my_app, :partitioned_cache, :command, :stop]
	[:my_app, :partitioned_cache, :primary, :command, :stop]
	[:my_app, :partitioned_cache, :command, :exception]
	[:my_app, :partitioned_cache, :primary, :command, :exception]

As you may notice, the telemetry prefix by default for the partitioned cache
is [:my_app, :partitioned_cache], and the prefix for its primary storage
[:my_app, :partitioned_cache, :primary].
See also the Telemetry guide
for more information and examples.

  
    
  
  Adapter-specific telemetry events


This adapter exposes following Telemetry events:
	telemetry_prefix ++ [:bootstrap, :started] - Dispatched by the adapter
when the bootstrap process is started.
	Measurements: %{system_time: non_neg_integer}

	Metadata:
%{
  adapter_meta: %{optional(atom) => term},
  cluster_nodes: [node]
}



	telemetry_prefix ++ [:bootstrap, :stopped] - Dispatched by the adapter
when the bootstrap process is stopped.
	Measurements: %{system_time: non_neg_integer}

	Metadata:
%{
  adapter_meta: %{optional(atom) => term},
  cluster_nodes: [node],
  reason: term
}



	telemetry_prefix ++ [:bootstrap, :exit] - Dispatched by the adapter
when the bootstrap has received an exit signal.
	Measurements: %{system_time: non_neg_integer}

	Metadata:
%{
  adapter_meta: %{optional(atom) => term},
  cluster_nodes: [node],
  reason: term
}



	telemetry_prefix ++ [:bootstrap, :joined] - Dispatched by the adapter
when the bootstrap has joined the cache to the cluster.
	Measurements: %{system_time: non_neg_integer}

	Metadata:
%{
  adapter_meta: %{optional(atom) => term},
  cluster_nodes: [node]
}





  
    
  
  Stats


This adapter depends on the primary storage adapter for the stats support.
Therefore, it is important to ensure the underlying primary storage adapter
does support stats, otherwise, you may get unexpected errors.

  
    
  
  Extended API


This adapter provides some additional convenience functions to the
Nebulex.Cache API.
Retrieving the primary storage or local cache module:
MyCache.__primary__()
Retrieving the cluster nodes associated with the given cache name:
MyCache.nodes()
Get a cluster node based on the given key:
MyCache.get_node("mykey")
Joining the cache to the cluster:
MyCache.join_cluster()
Leaving the cluster (removes the cache from the cluster):
MyCache.leave_cluster()

  
    
  
  Caveats of partitioned adapter


For Nebulex.Cache.get_and_update/3 and Nebulex.Cache.update/4,
they both have a parameter that is the anonymous function, and it is compiled
into the module where it is created, which means it necessarily doesn't exists
on remote nodes. To ensure they work as expected, you must provide functions
from modules existing in all nodes of the group.

Built-in adapter for replicated cache topology.

  
    
  
  Overall features


	Replicated cache topology.
	Configurable primary storage adapter.
	Cache-level locking when deleting all entries or adding new nodes.
	Key-level (or entry-level) locking for key-based write-like operations.
	Support for transactions via Erlang global name registration facility.
	Stats support rely on the primary storage adapter.


  
    
  
  Replicated Cache Topology


A replicated cache is a clustered, fault tolerant cache where data is fully
replicated to every member in the cluster. This cache offers the fastest read
performance with linear performance scalability for reads but poor scalability
for writes (as writes must be processed by every member in the cluster).
Because data is replicated to all servers, adding servers does not increase
aggregate cache capacity.
There are several challenges to building a reliably replicated cache. The
first is how to get it to scale and perform well. Updates to the cache have
to be sent to all cluster nodes, and all cluster nodes have to end up with
the same data, even if multiple updates to the same piece of data occur at
the same time. Also, if a cluster node requests a lock, ideally it should
not have to get all cluster nodes to agree on the lock or at least do it in
a very efficient way (:global is used here), otherwise it will scale
extremely poorly; yet in the case of a cluster node failure, all of the data
and lock information must be kept safely.
The best part of a replicated cache is its access speed. Since the data is
replicated to each cluster node, it is available for use without any waiting.
This is referred to as "zero latency access," and is perfect for situations
in which an application requires the highest possible speed in its data
access.
However, there are some limitations:
	Cost Per Update - Updating a replicated cache requires pushing
the new version of the data to all other cluster members, which will
limit scalability if there is a high frequency of updates per member.

	Cost Per Entry - The data is replicated to every cluster member,
so Memory Heap space is used on each member, which will impact
performance for large caches.


Based on "Distributed Caching Essential Lessons" by Cameron Purdy.


  
    
  
  Usage


When used, the Cache expects the :otp_app and :adapter as options.
The :otp_app should point to an OTP application that has the cache
configuration. For example:
defmodule MyApp.ReplicatedCache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Replicated
end
Optionally, you can configure the desired primary storage adapter with the
option :primary_storage_adapter; defaults to Nebulex.Adapters.Local.
defmodule MyApp.ReplicatedCache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Replicated,
    primary_storage_adapter: Nebulex.Adapters.Local
end
The configuration for the cache must be in your application environment,
usually defined in your config/config.exs:
config :my_app, MyApp.ReplicatedCache,
  primary: [
    gc_interval: 3_600_000,
    backend: :shards
  ]
If your application was generated with a supervisor (by passing --sup
to mix new) you will have a lib/my_app/application.ex file containing
the application start callback that defines and starts your supervisor.
You just need to edit the start/2 function to start the cache as a
supervisor on your application's supervisor:
def start(_type, _args) do
  children = [
    {MyApp.ReplicatedCache, []},
    ...
  ]
See Nebulex.Cache for more information.

  
    
  
  Options


This adapter supports the following options and all of them can be given via
the cache configuration:
	:primary - The options that will be passed to the adapter associated
with the local primary storage. These options will depend on the local
adapter to use.

	:task_supervisor_opts - Start-time options passed to
Task.Supervisor.start_link/1 when the adapter is initialized.



  
    
  
  Shared options


Almost all of the cache functions outlined in Nebulex.Cache module
accept the following options:
	:timeout - The time-out value in milliseconds for the command that
will be executed. If the timeout is exceeded, then the current process
will exit. For executing a command on remote nodes, this adapter uses
Task.await/2 internally for receiving the result, so this option tells
how much time the adapter should wait for it. If the timeout is exceeded,
the task is shut down but the current process doesn't exit, only the
result associated with that task is skipped in the reduce phase.


  
    
  
  Telemetry events


This adapter emits all recommended Telemetry events, and documented
in Nebulex.Cache module (see "Adapter-specific events" section).
Since the replicated adapter depends on the configured primary storage
adapter (local cache adapter), this one may also emit Telemetry events.
Therefore, there will be events emitted by the replicated adapter as well
as the primary storage adapter. For example, for the cache defined before
MyApp.ReplicatedCache, these would be the emitted events:
	[:my_app, :replicated_cache, :command, :start]
	[:my_app, :replicated_cache, :primary, :command, :start]
	[:my_app, :replicated_cache, :command, :stop]
	[:my_app, :replicated_cache, :primary, :command, :stop]
	[:my_app, :replicated_cache, :command, :exception]
	[:my_app, :replicated_cache, :primary, :command, :exception]

As you may notice, the telemetry prefix by default for the replicated cache
is [:my_app, :replicated_cache], and the prefix for its primary storage
[:my_app, :replicated_cache, :primary].
See also the Telemetry guide
for more information and examples.

  
    
  
  Stats


This adapter depends on the primary storage adapter for the stats support.
Therefore, it is important to ensure the underlying primary storage adapter
does support stats, otherwise, you may get unexpected errors.

  
    
  
  Extended API


This adapter provides some additional convenience functions to the
Nebulex.Cache API.
Retrieving the primary storage or local cache module:
MyCache.__primary__()
Retrieving the cluster nodes associated with the given cache name:
MyCache.nodes()
Joining the cache to the cluster:
MyCache.join_cluster()
Leaving the cluster (removes the cache from the cluster):
MyCache.leave_cluster()

  
    
  
  Adapter-specific telemetry events


This adapter exposes following Telemetry events:
	telemetry_prefix ++ [:replication] - Dispatched by the adapter
when a replication error occurs due to a write-like operation
under-the-hood.
	Measurements: %{rpc_errors: non_neg_integer}

	Metadata:
%{
  adapter_meta: %{optional(atom) => term},
  rpc_errors: [{node, error :: term}]
}



	telemetry_prefix ++ [:bootstrap] - Dispatched by the adapter at start
time when there are errors while syncing up with the cluster nodes.
	Measurements:
%{
  failed_nodes: non_neg_integer,
  remote_errors: non_neg_integer
}

	Metadata:
%{
  adapter_meta: %{optional(atom) => term},
  failed_nodes: [node],
  remote_errors: [term]
}





  
    
  
  Caveats of replicated adapter


As it is explained in the beginning, a replicated topology not only brings
with advantages (mostly for reads) but also with some limitations and
challenges.
This adapter uses global locks (via :global) for all operation that modify
or alter the cache somehow to ensure as much consistency as possible across
all members of the cluster. These locks may be per key or for the entire cache
depending on the operation taking place. For that reason, it is very important
to be aware about those operation that can potentially lead to performance and
scalability issues, so that you can do a better usage of the replicated
adapter. The following is with the operations and aspects you should pay
attention to:
	Starting and joining a new replicated node to the cluster is the most
expensive action, because all write-like operations across all members of
the cluster are blocked until the new node completes the synchronization
process, which involves copying cached data from any of the existing
cluster nodes into the new node, and this could be very expensive
depending on the number of caches entries. For that reason, adding new
nodes is considered an expensive operation that should happen only from
time to time.

	Deleting all entries. When Nebulex.Cache.delete_all/2 action is
executed, like in the previous case, all write-like operations in all
members of the cluster are blocked until the deletion action is completed
(this implies deleting all cached data from all cluster nodes). Therefore,
deleting all entries from cache is also considered an expensive operation
that should happen only from time to time.

	Write-like operations based on a key only block operations related to
that key across all members of the cluster. This is not as critical as
the previous two cases but it is something to keep in mind anyway because
if there is a highly demanded key in terms of writes, that could be also
a potential bottleneck.


Summing up, the replicated cache topology along with this adapter should
be used mainly when the the reads clearly dominate over the writes (e.g.:
Reads 80% and Writes 20% or less). Besides, operations like deleting all
entries from cache or adding new nodes must be executed only once in a while
to avoid performance issues, since they are very expensive.

Cache's main interface; defines the cache abstraction layer which is
highly inspired by Ecto.
A Cache maps to an underlying implementation, controlled by the
adapter. For example, Nebulex ships with a default adapter that
implements a local generational cache.
When used, the Cache expects the :otp_app and :adapter as options.
The :otp_app should point to an OTP application that has the cache
configuration. For example, the Cache:
defmodule MyApp.Cache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Local
end
Could be configured with:
config :my_app, MyApp.Cache,
  backend: :shards,
  gc_interval: :timer.hours(12),
  max_size: 1_000_000,
  allocated_memory: 2_000_000_000,
  gc_cleanup_min_timeout: :timer.seconds(10),
  gc_cleanup_max_timeout: :timer.minutes(10)
Most of the configuration that goes into the config is specific
to the adapter. For this particular example, you can check
Nebulex.Adapters.Local
for more information. In spite of this, the following configuration values
are shared across all adapters:
	:name - The name of the Cache supervisor process.

	:telemetry_prefix - It is recommend for adapters to publish events
using the Telemetry library. By default, the telemetry prefix is based
on the module name, so if your module is called MyApp.Cache, the prefix
will be [:my_app, :cache]. See the "Telemetry events" section to see
what events recommended for the  adapters to publish.. Note that if you
have multiple caches, you should keep the :telemetry_prefix consistent
for each of them and use the :cache and/or :name (in case of a named
or dynamic cache) properties in the event metadata for distinguishing
between caches.

	:telemetry - An optional flag to tell the adapters whether Telemetry
events should be emitted or not. Defaults to true.

	:stats - Boolean to define whether or not the cache will provide stats.
Defaults to false. Each adapter is responsible for providing stats by
implementing Nebulex.Adapter.Stats behaviour. See the "Stats" section
below.



  
    
  
  Telemetry events


Similar to Ecto or Phoenix, Nebulex also provides built-in Telemetry events
applied to all caches, and cache adapter-specific events.

  
    
  
  Nebulex built-in events


The following events are emitted by all Nebulex caches:
	[:nebulex, :cache, :init] - it is dispatched whenever a cache starts.
The measurement is a single system_time entry in native unit. The
metadata is the :cache and all initialization options under :opts.


  
    
  
  Adapter-specific events


It is recommend the adapters to publish certain Telemetry events listed
below. Those events will use the :telemetry_prefix outlined above which
defaults to [:my_app, :cache].
For instance, to receive all events published by a cache called MyApp.Cache,
one could define a module:
defmodule MyApp.Telemetry do
  def handle_event(
        [:my_app, :cache, :command, event],
        measurements,
        metadata,
        config
      ) do
    case event do
      :start ->
        # Handle start event ...

      :stop ->
        # Handle stop event ...

      :exception ->
        # Handle exception event ...
    end
  end
end
Then, in the Application.start/2 callback, attach the handler to this event
using a unique handler id:
:telemetry.attach(
  "my-app-handler-id",
  [:my_app, :cache, :command],
  &MyApp.Telemetry.handle_event/4,
  %{}
)
See the telemetry documentation
for more information.
The following are the events you should expect from Nebulex. All examples
below consider a cache named MyApp.Cache:
[:my_app, :cache, :command, :start]
This event should be invoked on every cache call sent to the adapter before
the command logic is executed.
The :measurements map will include the following:
	:system_time - The current system time in native units from calling:
System.system_time().

A Telemetry :metadata map including the following fields. Each cache adapter
may emit different information here. For built-in adapters, it will contain:
	:adapter_meta - The adapter metadata.
	:function_name - The name of the invoked adapter function.
	:args - The arguments of the invoked adapter function, omitting the
first argument, since it is the adapter metadata already included into
the event's metadata.

[:my_app, :cache, :command, :stop]
This event should be invoked on every cache call sent to the adapter after
the command logic is executed.
The :measurements map will include the following:
	:duration - The time spent executing the cache command. The measurement
is given in the :native time unit. You can read more about it in the
docs for System.convert_time_unit/3.

A Telemetry :metadata map including the following fields. Each cache adapter
may emit different information here. For built-in adapters, it will contain:
	:adapter_meta - The adapter metadata.
	:function_name - The name of the invoked adapter function.
	:args - The arguments of the invoked adapter function, omitting the
first argument, since it is the adapter metadata already included into
the event's metadata.
	:result - The command result.

[:my_app, :cache, :command, :exception]
This event should be invoked when an error or exception occurs while executing
the cache command.
The :measurements map will include the following:
	:duration - The time spent executing the cache command. The measurement
is given in the :native time unit. You can read more about it in the
docs for System.convert_time_unit/3.

A Telemetry :metadata map including the following fields. Each cache adapter
may emit different information here. For built-in adapters, it will contain:
	:adapter_meta - The adapter metadata.
	:function_name - The name of the invoked adapter function.
	:args - The arguments of the invoked adapter function, omitting the
first argument, since it is the adapter metadata already included into
the event's metadata.
	:kind - The type of the error: :error, :exit, or :throw.
	:reason - The reason of the error.
	:stacktrace - The stacktrace.

NOTE: The events outlined above are the recommended for the adapters
to dispatch. However, it is highly recommended to review the used adapter
documentation to ensure it is fully compatible with these events, perhaps
differences, or perhaps also additional events.

  
    
  
  Stats


Stats are provided by the adapters by implementing the optional behaviour
Nebulex.Adapter.Stats. This behaviour exposes a callback to return the
current cache stats.  Nevertheless, the behaviour brings with a default
implementation using Erlang counters, which is used by the
local built-in adapter (Nebulex.Adapters.Local).
One can enable the stats by setting the option :stats to true.
For example, in the configuration file:
config :my_app, MyApp.Cache,
  stats: true,
  ...
Remember to check if the underlying adapter implements the
Nebulex.Adapter.Stats behaviour.

See Nebulex.Cache.stats/0 for more information.

  
    
  
  Dispatching stats via Telemetry


It is possible to emit Telemetry events for the current stats via
Nebulex.Cache.dispatch_stats/1, but it has to be invoked explicitly;
Nebulex does not emit this Telemetry event automatically. But it is very
easy to emit this event using :telemetry_poller.
For example, one can define a custom pollable measurement:
:telemetry_poller.start_link(
  measurements: [
    {MyApp.Cache, :dispatch_stats, []},
  ],
  # configure sampling period - default is :timer.seconds(5)
  period: :timer.seconds(30),
  name: :my_cache_stats_poller
)
Or you can also start the :telemetry_poller process along with your
application supervision tree:
def start(_type, _args) do
  my_cache_stats_poller_opts = [
    measurements: [
      {MyApp.Cache, :dispatch_stats, []},
    ],
    period: :timer.seconds(30),
    name: :my_cache_stats_poller
  ]

  children = [
    {MyApp.Cache, []},
    {:telemetry_poller, my_cache_stats_poller_opts}
  ]

  opts = [strategy: :one_for_one, name: MyApp.Supervisor]
  Supervisor.start_link(children, opts)
end
See Nebulex Telemetry Guide
for more information.

  
    
  
  Distributed topologies


Nebulex provides the following adapters for distributed topologies:
	Nebulex.Adapters.Partitioned - Partitioned cache topology.
	Nebulex.Adapters.Replicated - Replicated cache topology.
	Nebulex.Adapters.Multilevel - Multi-level distributed cache topology.

These adapters work more as wrappers for an existing local adapter and provide
the distributed topology on top of it. Optionally, you can set the adapter for
the primary cache storage with the option :primary_storage_adapter. Defaults
to Nebulex.Adapters.Local.

At its core, the abstraction applies caching to Elixir functions, reducing
thus the number of executions based on the information available in the
cache. That is, each time a targeted function is invoked, the abstraction
will apply a caching behavior checking whether the function has been already
executed and its result cached for the given arguments. If it has, then the
cached result is returned without having to execute the actual function;
if it has not, then function is executed, the result cached and returned
to the user so that, the next time the method is invoked, the cached result
is returned. This way, expensive functions (whether CPU or IO bound) can be
executed only once for a given set of parameters and the result reused
without having to actually execute the function again. The caching logic
is applied transparently without any interference to the invoker.
See Nebulex.Caching.Decorators for more information about
"Declarative annotation-based caching".

Declarative annotation-based caching via function
decorators.
For caching declaration, the abstraction provides three Elixir function
decorators: cacheable, cache_evict, and cache_put, which allow
functions to trigger cache population or cache eviction.
Let us take a closer look at each annotation.
Inspired by Spring Cache Abstraction.


  
    
  
  cacheable decorator


As the name implies, cacheable is used to demarcate functions that are
cacheable - that is, functions for whom the result is stored into the cache
so, on subsequent invocations (with the same arguments), the value in the
cache is returned without having to actually execute the function. In its
simplest form, the decorator/annotation declaration requires the name of
the cache associated with the annotated function:
@decorate cacheable(cache: Cache)
def get_account(id) do
  # the logic for retrieving the account ...
end
In the snippet above, the function get_account/1 is associated with the
cache named Cache. Each time the function is called, the cache is checked
to see whether the invocation has been already executed and does not have
to be repeated.

  
    
  
  Default Key Generation


Since caches are essentially key-value stores, each invocation of a cached
function needs to be translated into a suitable key for cache access.
Out of the box, the caching abstraction uses a simple key-generator
based on the following algorithm:
	If no params are given, return 0.
	If only one param is given, return that param as key.
	If more than one param is given, return a key computed from the hashes
of all parameters (:erlang.phash2(args)).

IMPORTANT: Since Nebulex v2.1.0, the default key generation implements
the algorithm described above, breaking backward compatibility with older
versions. Therefore, you may need to change your code in case of using the
default key generation.

The default key generator is provided by the cache via the callback
Nebulex.Cache.__default_key_generator__/0 and it is applied only
if the option key: or keys: is not configured. Defaults to
Nebulex.Caching.SimpleKeyGenerator. You can change the default
key generator at compile time with the option :default_key_generator.
For example, one can define a cache with a default key generator as:
defmodule MyApp.Cache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Local,
    default_key_generator: __MODULE__

  @behaviour Nebulex.Caching.KeyGenerator

  @impl true
  def generate(mod, fun, args), do: :erlang.phash2({mod, fun, args})
end
The key generator module must implement the Nebulex.Caching.KeyGenerator
behaviour.
IMPORTANT: There are some caveats to keep in mind when using
the key generator, therefore, it is highly recommended to review
Nebulex.Caching.KeyGenerator behaviour documentation before.

Also, you can provide a different key generator at any time
(overriding the default one) when using any caching annotation
through the option :key_generator. For example:
# With a module implementing the key-generator behaviour
@decorate cache_put(cache: Cache, key_generator: CustomKeyGenerator)
def update_account(account) do
  # the logic for updating the given entity ...
end

# With the shorthand tuple {module, args}
@decorate cache_put(
            cache: Cache,
            key_generator: {CustomKeyGenerator, [account.name]}
          )
def update_account2(account) do
  # the logic for updating the given entity ...
end

# With a MFA tuple
@decorate cache_put(
            cache: Cache,
            key_generator: {AnotherModule, :genkey, [account.id]}
          )
def update_account3(account) do
  # the logic for updating the given entity ...
end
The :key_generator option is available for all caching annotations.


  
    
  
  Custom Key Generation Declaration


Since caching is generic, it is quite likely the target functions have
various signatures that cannot be simply mapped on top of the cache
structure. This tends to become obvious when the target function has
multiple arguments out of which only some are suitable for caching
(while the rest are used only by the function logic). For example:
@decorate cacheable(cache: Cache)
def get_account(email, include_users?) do
  # the logic for retrieving the account ...
end
At first glance, while the boolean argument influences the way the account
is found, it is no use for the cache.
For such cases, the cacheable decorator allows the user to specify the
key explicitly based on the function attributes.
@decorate cacheable(cache: Cache, key: {Account, email})
def get_account(email, include_users?) do
  # the logic for retrieving the account ...
end

@decorate cacheable(cache: Cache, key: {Account, user.account_id})
def get_user_account(%User{} = user) do
  # the logic for retrieving the account ...
end
It is also possible passing options to the cache, like so:
@decorate cacheable(
            cache: Cache,
            key: {Account, email},
            opts: [ttl: 300_000]
          )
def get_account(email, include_users?) do
  # the logic for retrieving the account ...
end
See the "Shared Options" section below.

  
    
  
  Functions with multiple clauses


Since decorator lib
is used, it is important to be aware of its recommendations, warns,
limitations, and so on. In this case, for functions with multiple clauses
the general advice is to create an empty function head, and call the
decorator on that head, like so:
@decorate cacheable(cache: Cache, key: email)
def get_account(email \\ nil)

def get_account(nil), do: nil

def get_account(email) do
  # the logic for retrieving the account ...
end

  
    
  
  cache_put decorator


For cases where the cache needs to be updated without interfering with the
function execution, one can use the cache_put decorator. That is, the
method will always be executed and its result placed into the cache
(according to the cache_put options). It supports the same options as
cacheable.
@decorate cache_put(cache: Cache, key: {Account, acct.email})
def update_account(%Account{} = acct, attrs) do
  # the logic for updating the account ...
end
Note that using cache_put and cacheable annotations on the same function
is generally discouraged because they have different behaviors. While the
latter causes the method execution to be skipped by using the cache, the
former forces the execution in order to execute a cache update. This leads
to unexpected behavior and with the exception of specific corner-cases
(such as decorators having conditions that exclude them from each other),
such declarations should be avoided.

  
    
  
  cache_evict decorator


The cache abstraction allows not just the population of a cache store but
also eviction. This process is useful for removing stale or unused data from
the cache. Opposed to cacheable, the decorator cache_evict demarcates
functions that perform cache eviction, which are functions that act as
triggers for removing data from the cache. The cache_evict decorator not
only allows a key to be specified, but also a set of keys. Besides, extra
options likeall_entries which indicates whether a cache-wide eviction
needs to be performed rather than just an entry one (based on the key or
keys):
@decorate cache_evict(cache: Cache, key: {Account, email})
def delete_account_by_email(email) do
  # the logic for deleting the account ...
end

@decorate cacheable(
            cache: Cache,
            keys: [{Account, acct.id}, {Account, acct.email}]
          )
def delete_account(%Account{} = acct) do
  # the logic for deleting the account ...
end

@decorate cacheable(cache: Cache, all_entries: true)
def delete_all_accounts do
  # the logic for deleting all the accounts ...
end
The option all_entries: comes in handy when an entire cache region needs
to be cleared out - rather than evicting each entry (which would take a
long time since it is inefficient), all the entries are removed in one
operation as shown above.

  
    
  
  Shared Options


All three cache annotations explained previously accept the following
options:
	:cache - Defines what cache to use (required). Raises ArgumentError
if the option is not present. It can be also a MFA tuple to resolve the
cache dynamically in runtime by calling it. See "The :cache option"
section below for more information.

	:key - Defines the cache access key (optional). It overrides the
:key_generator option. If this option is not present, a default
key is generated by the configured or default key generator.

	:opts - Defines the cache options that will be passed as argument
to the invoked cache function (optional).

	:match - Match function match_fun/0. This function is for matching
and deciding whether the code-block evaluation result (which is received
as an argument) is cached or not. The function should return:
	true - the code-block evaluation result is cached as it is
(the default).
	{true, value} - value is cached. This is useful to set what
exactly must be cached.
	{true, value, opts} - value is cached with the options given by
opts. This return allows us to set the value to be cached, as well
as the runtime options for storing it (e.g.: the ttl).
	false - Nothing is cached.

The default match function looks like this:
fn
  {:error, _} -> false
  :error -> false
  nil -> false
  _ -> true
end
By default, if the code-block evaluation returns any of the following
terms/values nil, :error, {:error, term}, the default match
function returns false (the returned result is not cached),
otherwise, true is returned (the returned result is cached).

	:key_generator - The custom key-generator to be used (optional).
If present, this option overrides the default key generator provided
by the cache, and it is applied only if the option key: or keys:
is not present. In other words, the option key: or keys: overrides
the :key_generator option. See "The :key_generator option" section
below for more information about the possible values.

	:on_error - It may be one of :raise (the default) or :nothing.
The decorators/annotations call the cache under the hood, hence,
by default, any error or exception at executing a cache command
is propagated. When this option is set to :nothing, any error
or exception executing a cache command is ignored and the annotated
function is executed normally.



  
    
  
  The :cache option


The cache option can be the de defined cache module or an MFA tuple to
resolve the cache dynamically in runtime. When it is an MFA tuple, the
MFA is invoked passing the calling module, function name, and arguments
by default, and the MFA arguments are passed as extra arguments.
For example:
@decorate cacheable(cache: {MyApp.Cache, :cache, []}, key: var)
def some_function(var) do
  # Some logic ...
end
The annotated function above will call MyApp.Cache.cache(mod, fun, args)
to resolve the cache in runtime, where mod is the calling module, fun
the calling function name, and args the calling arguments.
Also, we can define the function passing some extra arguments, like so:
@decorate cacheable(cache: {MyApp.Cache, :cache, ["extra"]}, key: var)
def some_function(var) do
  # Some logic ...
end
In this case, the MFA will be invoked by adding the extra arguments, like:
MyApp.Cache.cache(mod, fun, args, "extra").

  
    
  
  The :key_generator option


The possible values for the :key_generator are:
	A module implementing the Nebulex.Caching.KeyGenerator behaviour.

	A MFA tuple {module, function, args} for a function to call to
generate the key before the cache is invoked. A shorthand value of
{module, args} is equivalent to
{module, :generate, [calling_module, calling_function_name, args]}.



  
    
  
  Putting all together


Supposing we are using Ecto and we want to define some cacheable functions
within the context MyApp.Accounts:
# The config
config :my_app, MyApp.Cache,
  gc_interval: 86_400_000, #=> 1 day
  backend: :shards

# The Cache
defmodule MyApp.Cache do
  use Nebulex.Cache,
    otp_app: :my_app,
    adapter: Nebulex.Adapters.Local
end

# Some Ecto schema
defmodule MyApp.Accounts.User do
  use Ecto.Schema

  schema "users" do
    field(:username, :string)
    field(:password, :string)
    field(:role, :string)
  end

  def changeset(user, attrs) do
    user
    |> cast(attrs, [:username, :password, :role])
    |> validate_required([:username, :password, :role])
  end
end

# Accounts context
defmodule MyApp.Accounts do
  use Nebulex.Caching

  alias MyApp.Accounts.User
  alias MyApp.{Cache, Repo}

  @ttl :timer.hours(1)

  @decorate cacheable(cache: Cache, key: {User, id}, opts: [ttl: @ttl])
  def get_user!(id) do
    Repo.get!(User, id)
  end

  @decorate cacheable(
              cache: Cache,
              key: {User, username},
              opts: [ttl: @ttl]
            )
  def get_user_by_username(username) do
    Repo.get_by(User, [username: username])
  end

  @decorate cache_put(
              cache: Cache,
              keys: [{User, usr.id}, {User, usr.username}],
              match: &match_update/1
            )
  def update_user(%User{} = usr, attrs) do
    usr
    |> User.changeset(attrs)
    |> Repo.update()
  end

  defp match_update({:ok, usr}), do: {true, usr}
  defp match_update({:error, _}), do: false

  @decorate cache_evict(
              cache: Cache,
              keys: [{User, usr.id}, {User, usr.username}]
            )
  def delete_user(%User{} = usr) do
    Repo.delete(usr)
  end

  def create_user(attrs \\ %{}) do
    %User{}
    |> User.changeset(attrs)
    |> Repo.insert()
  end
end
See Cache Usage Patterns Guide.

Cache key generator. Used for creating a key based on the given module,
function name and its arguments (the module and function name are used
as context).
See the default implementation Nebulex.Caching.SimpleKeyGenerator.

  
    
  
  Caveats when using the key generator


Since the callback generate/3 is invoked passing the calling module where
the annotated function is defined, the name of the annotated function, and the
arguments given to that annotated function, there are some caveats to keep in
mind:
	Only arguments explicitly assigned to a variable will be included when
calling the callback generate/3.
	Ignored or underscored arguments will be ignored.
	Pattern-matching expressions without a variable assignment will be
ignored. If there is a pattern-matching, it has to be explicitly
assigned to a variable so it can be included when calling the
callback generate/3.

For example, suppose you have a module with an annotated function:
defmodule MyApp.SomeModule do
  use Nebulex.Caching

  alias MyApp.{Cache, CustomKeyGenerator}

  @decorate cacheable(cache: Cache, key_generator: CustomKeyGenerator)
  def get_something(x, _ignored, _, {_, _}, [_, _], %{a: a}, %{} = y) do
    # Function's logic
  end
end
The generator will be invoked like so:
MyKeyGenerator.generate(MyApp.SomeModule, :get_something, [x, y])
Based on the caveats described above, only the arguments x and y are
included when calling the callback generate/3.

Default key generator implementation.
It implementats a simple algorithm:
	If no params are given, return 0.
	If only one param is given, return that param as key.
	If more than one param is given, return a key computed from the hashes
of all parameters (:erlang.phash2(args)).

Based on the default key generation in Spring Cache Abstraction.

This implementation aims to cover those simple/generic scenarios where the
key generated based on the arguments only, fulfill the needs. For example:
defmodule MyApp.Users do
  use Nebulex.Caching

  alias MayApp.Cache

  @decorate cacheable(cache: Cache)
  def get_user(id) do
    # logic for retrieving a user...
  end

  @decorate cache_evict(cache: Cache)
  def delete_user(id) do
    # logic for deleting a user...
  end
end
The key generator will generate the same key for both, cacheable and
evict functions; since it is ge